code:
;-------------------------------------------------
;EV9 インキュバスの扉
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_11_EVENT_20
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_SLOT_NAME = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_11_EVENT_20
#DIM EVENT_FLAG
#DIM TARGET_ID
#DIM  DYNAMIC CHARA_ID
#DIM 口上カラー_カール
#DIM 口上カラー_カーラ
#DIM 評定額

CHARA_ID = NAME_TO_CHARA("インキュバス・カール")
IF CHARA_ID == -1
    CALL QST_QUEST11_CREATE_UNIQUE_MOB1
    CHARA_ID = RESULT
ENDIF
口上カラー_カール = CFLAG:CHARA_ID:口上カラーコード
CFLAG:CHARA_ID:面識 = 1

CHARA_ID = NAME_TO_CHARA("サキュバス・カーラ")
IF CHARA_ID == -1
    CALL QST_QUEST11_CREATE_UNIQUE_MOB2
    CHARA_ID = RESULT
ENDIF
口上カラー_カーラ = CFLAG:CHARA_ID:口上カラーコード
CFLAG:CHARA_ID:面識 = 1

;初回
IF !QST_SV_QUEST_FLAG:QST_有効クエストID
    PRINTFORML 長い回廊を抜け、%QST_LEADER_NAME()%は突き当りの扉に立った。恐らくここが最後の部屋であろう
    PRINTFORMW 意を決して扉を開けるとまず目の前に飛び込んできたのは乱交と痴態に耽る無数の肉のうねり……を模した像の数々だ
    PRINTL
    PRINTFORMW 交わり、絡み合う肢体の彫像の山の上には少年と少女が腰掛けている……
    PRINTFORMW 一見はあたかも双子のようだ。ただし少年の方は赤褐色の瞳、少女の方は冷たい青い瞳をしている
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「おっ、ここまで来られた人間は本当に久しぶりだよ。ようこそ淫魔の間へ」@/C@", 1)
    PRINTL
    PRINTFORMW 少年が気さくに語りかけてくる
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「あはは、アナタが頑張ってるところここからじっくり見させてもらってたよ！」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「必死になってるニンゲンって……やっぱり見てて本当に面白いね！」」@/C@", 1)
    PRINTW
    PRINTFORML 少女の小馬鹿にしたような態度に%QST_LEADER_NAME()%は苛立ちを覚える
    PRINTFORMW 見た目こそ人間の子どもだが、背中から生えた翼や容姿から、二人は恐らく魔族、淫魔の類だろう……
    PRINTL
    PRINTFORMW 同時に……酷く恐ろしい悪寒が背筋を伝うのを%QST_LEADER_NAME()%は感じた
    PRINTL
    PRINTFORMW 直感は、この二人と剣を交えたところで絶対に勝利し得ないことを告げている……
    PRINTL
    CALL PRINT_TITLE("双子の淫魔", "イベント")
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「ふ～ん。結構勘はいいんだね。僕はカール。インキュバスのカールだ」@/C@", 1)
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「こっちはサキュバスのカーラ。僕たちはこの迷宮の主でね、二人とも昔は結構崇められたり、奉られたりしたんだよ」@/C@", 1)
    PRINTW
    PRINTFORML インキュバス、サキュバス二人の淫魔は軽やかな足取りで間合いを詰めてくる
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「昔はニンゲンが色々と捧げてくれたんだよね～。可愛い娘とか男の子とか」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「その代わりに私たちはちょっと魔力を分けてあげたり、ニンゲンがありがたがっている石ころをあげたりしたんだ」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「その辺に転がってるのはみ～んなその名残。絞り尽くしちゃって、イケニエちゃんはみんな空っぽの石ころになっちゃった」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「最後にイケニエちゃん貰ってからもう何百年になるのかな～、最近また少し賑やかになってきたけど……長い間すっごく暇だったんだよ！」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「さて、そんなわけで人間！取引といこう」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「僕たちは別に今君を取って食おうとは思っていない」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「ただ君がここに来るまでに味わい、散々溜め込んできた\"欲\"をちょっと分けてくれさえすればいい」@/C@", 1)
    WAIT
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「代わりに僕らは君の役に立つものを提供しよう。どうだい？悪い取引ではないだろう？」@/C@", 1)
    WAIT
    PRINTL
    PRINTFORML 淫魔たちの提案は、実質的に選択肢が無いように思われた
    PRINTFORMW 断れば……恐らくただでは済まないだろう。さて、どうしたものか
    QST_SV_QUEST_FLAG:QST_有効クエストID = 1
ELSE
    PRINTFORML 長い回廊を抜け、%QST_LEADER_NAME()%は突き当りの扉に立った。扉の向こうからは並々ならぬ気配がする
    PRINTFORMW 意を決して再び扉を開けると二人の淫魔が挑発的な笑みを浮かべてこちらを眺めていた……
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「あは！また遊びに来てくれたんだね！ぐちゃぐちゃにされるのが癖になっちゃったのかな？可愛いね！」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「まぁ楽しんでくれているなら僕たちも苦労した甲斐があったというものだよ」@/C@", 1)
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「さて、それじゃあ取引と行こうか。もちろん君の色欲を分けて……くれるんだよね？」@/C@", 1)
ENDIF

PRINTL
CALL ASK_YN("取引を受ける", "抵抗する")
PRINTL

IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%が渋々頷くとカールと名乗る淫魔はにこりと微笑んだ
    PRINTL
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うんうん、話が早くて助かるよ。君は中々活きがいいからね。僕たちとしても大事にしたいんだ」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「釣りは釣果よりも、釣る行為自体が面白いだろう？だから、長く楽しめるようにしないとね」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「キャッチアンドリリースってやつだね！」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「それじゃあ早速欲情を分けてもらおうか……、どれどれ……」@/C@", 1)
    PRINTL
    CALL PRINT_TITLE("淫魔の品定め", "エロイベント")
    PRINTW
    CALL QST_QUEST_11_EVENT_20_淫魔の評定
    評定額 = RESULT
    PRINTL
    PRINTFORMW …………
    SELECTCASE 評定額
        CASE IS >= 100
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うんうん！イイね！すごくいい、本当に最上級のエロ雌豚だ！」@/C@", 1)
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「君のみっともなくてどうしようもない痴態を眺めてるだけでも、僕たちはもうしばらくやっていけそうだよ！」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「エロ雌豚ちゃんはもう頭の中真っピンクでエッチなことしか考えられないって感じだね！可愛いね！」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「豚ちゃんはきっと普通にエッチするだけじゃ満足できない子だろうから、またいつでも遊びに来てね！」@/C@", 1)
            WAIT
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「その時はもしかしたら～、今度は私達が直接遊んであげるかも！」@/C@", 1)
            PRINTW
        CASE IS >= 30
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うん。まぁ、まぁまぁかな。中の上くらいの豚」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「お姉さんはぁ……ちょ～っと残念かな。もっとアナタがヒィヒィ言いながらエロ汁垂れ流して走り回ってるところ見たかったんだけどなぁ～」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「でもちょっとは面白かったから、これ、あげるね！次はもうちょっとがんばってね！」@/C@", 1)
            PRINTW
        CASEELSE
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「はぁ……、あのさ……全然ダメじゃん。やる気あるの？なんていうかさ、不真面目すぎるんだよね……」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「そうだねぇ……カール。これは期待外れ……。ちょっと間違えて殺しちゃいそう」@/C@", 1)
            PRINTW
            CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「……一応は来てくれたお客さんだからこれあげるけどさ、次はもうちょっと、真面目にやってね」@/C@", 1)
            PRINTW
    ENDSELECT
    PRINTL
    PRINTFORMW 淫魔たちは%QST_LEADER_NAME()%を散々好き放題に罵倒した上で、約束通り報酬を手渡してきた……
    PRINTFORMW %QST_LEADER_NAME()%は羞恥に顔を赤らめ、歯ぎしりつつも、剣を交えずとも実感できる圧倒的な力量差を前に今はただ去ることしかできなかった……
    PRINTL
    PRINTFORMW 一先ず調査は完了した……。帰還するとしよう
    PRINTL
    CALL QST_CHANGE_ITEM(@"淫堕点,{MIN(評定額 / 10 * 10, 200)},")
    SELECTCASE 評定額
        CASE IS >= 100
            CALL QST_CHANGE_ITEM(@"{ITEM_金インゴット},1,")
            CALL QST_CHANGE_ITEM(@"{ITEM_銀インゴット},3,")
        CASE IS >= 30
            CALL QST_CHANGE_ITEM(@"{ITEM_銀インゴット},3,")
        CASEELSE
            CALL QST_CHANGE_ITEM(@"{ITEM_銀インゴット},1,")
    ENDSELECT
    TFLAG:クエスト_完了フラグ = 1
    QST_COOLTIME:QST_有効クエストID += 4
    QST_SV_QUEST_FLAG:QST_有効クエストID = 2
    CALL TRAIN_EXIT()

ELSE
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「……やれやれ、昔からまぁ馬鹿な連中だとは思ってたけど、何百年たってもちっとも変わらないね……」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「ねぇねぇカール、身の程知らずの家畜は～えっと、今風に言うなら……『わからせ』が必要だよね！」@/C@", 1)
    PRINTW
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「そうだね、カーラ。少し痛めつけて、己の立場をきちんとわきまえさせないといけないね」@/C@", 1)
    PRINTW
    PRINTFORMW 二体の淫魔がその身に秘めていた膨大な魔力を解き放つ……
    PRINTFORMW 空間がビリビリと震え、周囲の彫像が砕けて弾け飛ぶ。発狂しかねないほどの圧力と恐怖が%QST_LEADER_NAME()%を苛む
    PRINTL
    PRINTFORMW 解き放たれた膨大な魔力は、禍々しい黒馬の姿をとって顕現し、%QST_LEADER_NAME()%を包囲する
    PRINTL
    PRINTFORMW 状況は絶望的だ……。しかし後悔したところで最早手遅れだろう。圧力に負けて命までは手放さぬよう、死力を尽くして抗わねば……
    CALL CC_INIT()
    CALL CC_ADDCHARA("DEFENDER", NAME_TO_CHARA("インキュバス・カール"))
    CALL CC_ADDCHARA("DEFENDER", NAME_TO_CHARA("サキュバス・カーラ"))
    CALL CC_ADDMOB("DEFENDER", "ナイトメア１", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)
    CALL CC_ADDMOB("DEFENDER", "ナイトメア２", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)
    CALL CC_ADDMOB("DEFENDER", "ナイトメア３", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)
    CALL CC_ADDMOB("DEFENDER", "ナイトメア４", 3500, 3500, 45, 75 ,0, WPN_脅威選定_魔術師)

    FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
        CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
    NEXT
    CALL CLOSE_COMBAT()
    ;今のところ負けイベ
    IF RESULT == 2
        PRINTFORMW 激烈な魔力の奔流が炸裂し、%QST_LEADER_NAME()%は抗う余地もなく地面に叩きつけられた！
        PRINTL
        PRINTFORMW 朦朧とする意識の中、二人の淫魔が嘲笑の笑みを浮かべて自身を睥睨している
        PRINTL
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「ぷっ……アハハ！ダメじゃん！全然ダメじゃん！準備運動にもならないよ！」@/C@", 1)
        PRINTW
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「まぁ威勢の良さに免じて命は取らないでおいてあげるよ。でもね、聞き分けのない家畜は……ちゃんと躾けないとね」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「大丈夫？意識はある？これから好き勝手に虐めて絞らせてもらうけど、死なないように頑張ってね！」@/C@", 1)
        PRINTW
        CALL QST_QUEST_11_DEFEAT_EVENT_3()
    ELSE ;勝てないはず
        PRINTL
        PRINTFORMW %QST_LEADER_NAME()%は圧倒的な力で黒馬を屠り、淫魔を真正面からねじ伏せた……
        PRINTL
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「えっ、ちょっとまって？なんで？なんでこの状況で勝てるの？」@/C@", 1)
        PRINTW
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「困ったなぁ……君みたいな規格外なのが来ることは考えてなかったよ……」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「あーもう、参ったなぁ……こうなったときの段取り、なにも用意してないんだよね……」@/C@", 1)
        PRINTW
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「とりあえずさ、これあげるからさ……。今日は帰ってくれない？ごめんね！」@/C@", 1)
        PRINTW
        CALL QST_CHANGE_ITEM("名誉点,150,")
        PRINTFORMW 当初の目算が外れたのか、淫魔達はバツが悪そうに顔を見合わせると、瞬く間に姿を消してしまった
        PRINTL
        PRINTFORMW 一先ず調査は完了した……。帰還するとしよう
        PRINTL
        TFLAG:クエスト_完了フラグ = 1
        QST_COOLTIME:QST_有効クエストID += 4
        QST_SV_QUEST_FLAG:QST_有効クエストID = 2
        CALL TRAIN_EXIT()

    ENDIF
ENDIF

;-------------------------------------------------
; 探索の質を評価。ついでに罵倒される
;-------------------------------------------------
@QST_QUEST_11_EVENT_20_淫魔の評定
#DIM EVENT_FLAG
#DIM TARGET_ID
#DIM  DYNAMIC CHARA_ID
#DIM 口上カラー_カール
#DIM 口上カラー_カーラ
#DIM DYNAMIC ぶっかけ度
#DIM DYNAMIC 評定額

CHARA_ID = NAME_TO_CHARA("インキュバス・カール")
口上カラー_カール = CFLAG:CHARA_ID:口上カラーコード

CHARA_ID = NAME_TO_CHARA("サキュバス・カーラ")
口上カラー_カーラ = CFLAG:CHARA_ID:口上カラーコード

CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, 0))
;ぶっかけ具合
FOR LOCAL, 0, CUM_SLOT_NUM
    ぶっかけ度 += CUM_GET_AMOUNT:CHARA_ID:LOCAL
NEXT
ぶっかけ度 = ぶっかけ度 / 3
IF ぶっかけ度 > 35
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「うわぁ……身体中どろどろのザーメンまみれじゃん……。人間の雌豚の盛り方って昔から品がないないなぁ……」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「まぁ雌豚は雌豚らしく穴からザーメン垂れ流しながらアクメ顔晒してるくらいの方がそれらしくていいと思うよ」@/C@", 1)
    評定額 += 25
    PRINTW
ELSEIF ぶっかけ度 > 15
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「いい感じに身体から青臭い精臭撒き散らしてるね♪」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「香水なんかよりずっと魅力的だと思うよ！定期的にぶっかけてもらったらいいんじゃないかな！」@/C@", 1)
    評定額 += 10
    PRINTW
ENDIF
DEBUGPRINTFORML ぶっかけ度:{ぶっかけ度}
;露出_胸
IF C_COND("露出_胸:露出",, CHARA_ID)
    CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「あはっ！お洋服、破けちゃったんだ。可愛いおっぱい丸出しにしてここまで歩いてきたの？恥ずかしくないの？」@/C@", 1)
    評定額 += 25
    PRINTW
ENDIF
;露出_秘部
IF C_COND("露出_秘部:露出",, CHARA_ID)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「下着は……はいてないの？それともグチョグチョにしちゃったから脱いできたの？」@/C@", 1)
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「恥ずかしいところ丸見えじゃん。それってどうなのかなぁ……？」@/C@", 1)
    WAIT
    CALLF TAG_PRINT(@"@C:{口上カラー_カール}@「ニンゲンって、一応は社会性とかあるんだよね？」@/C@", 1)
    評定額 += 25
    PRINTW
ENDIF
;欲情LV
LOCAL = GET_PALAMLV(PALAM:(CHARA_ID):欲情)
SELECTCASE LOCAL
    CASE IS >= 12
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「大丈夫？息荒いよ？こんなに顔真っ赤にして、息も荒らげて……ただでさえミニマムな頭のなかを……スケベなことでいっぱいにしてるの？」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「うんうん、いいよ！すごくいいと思う！余計なことなんて考えないで、どうやって気持ちよくなるかだけ考えてる方が絶対楽しいよ！」@/C@", 1)
        PRINTW
    CASE IS >= 8
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「ふーん……。まだちょっと物足りないって顔してるね。まぁ余計なこと考えないで、もっと頭真っピンクにした方が楽しいと思うよ！」@/C@", 1)
        PRINTW
    CASEELSE
ENDSELECT
評定額 += LOCAL * 3
;潤滑LV
LOCAL:0 = GET_PALAMLV(PALAM:(CHARA_ID):潤滑)
SELECTCASE LOCAL
    CASE IS >= 9
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「アハハ！お股もヌルヌルでグショグショじゃん！きっといっぱいイッちゃったんだろうねぇ……」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「……どう、気持ちよかった？ねぇ、人様のおうちでメス汁ぽたぽた垂れ流して、汚しながら無様に喘いで、のたうち回って……、気持ちよかったんだよね？」@/C@", 1)
        PRINTW
    CASE IS >= 6
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「わぁ、お股びちょびちょだねぇ……。でもまだちょっと濡れ足りないのかな？」@/C@", 1)
        WAIT
        CALLF TAG_PRINT(@"@C:{口上カラー_カーラ}@「雌豚は潮吹いて浅ましくイキ散らかしてるくらいの方が、可愛げがあっていいと思うんだ！」@/C@", 1)
        PRINTW
    CASEELSE
ENDSELECT
評定額 += LOCAL * 3
RETURN 評定額

path: ERB/QUEST/クエストデータ/QUEST_11_淫魔の小迷宮/QUEST_11_淫魔の小迷宮_淫魔の間.ERB


code:
MAP_NAME = "淫魔の小迷宮"
MAP_IMAGE = "淫魔の小迷宮.webp"
PATH_COLOR = "0xff00"
PATH_ALPHA = "0x0"
NODE_BASE_COLOR = "0xEE1190"

repo_name: napo/erafl

path: ERB/QUEST/クエストデータ/QUEST_11_淫魔の小迷宮/淫魔の小迷宮.txt

license: unknown


code:
<DocumentElement>
  <GMAPDATA>
    <id>638145913543188710</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>入口</NODE_NAME>
    <POS_X>100</POS_X>
    <POS_Y>360</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913590166636</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>彫像の間</NODE_NAME>
    <POS_X>140</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>0,2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913635016064</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>200</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>1,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913734256553</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>大部屋</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>2,4,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913755294270</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>60</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913907045903</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>240</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>3,6,7,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913924505656</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>書斎</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>60</POS_Y>
    <PATH_LIST>5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145913945666133</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>140</POS_Y>
    <PATH_LIST>5,8,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145914087637586</id>
    <NODE_ID>8</NODE_ID>
    <NODE_NAME>通路</NODE_NAME>
    <POS_X>420</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>7,9,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638145914111958370</id>
    <NODE_ID>9</NODE_ID>
    <NODE_NAME>淫魔の間</NODE_NAME>
    <POS_X>500</POS_X>
    <POS_Y>300</POS_Y>
    <PATH_LIST>8,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>

path: ERB/QUEST/クエストデータ/QUEST_11_淫魔の小迷宮/淫魔の小迷宮.xml


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_12:野盗討伐
;-------------------------------------------------
@QST_TITLE_QUEST_12
RESULTS '= "野盗討伐"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_12
RESULTS '= "周期,探索,"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_12
RESULTS:0 '= "野盗の討伐はギルドに寄せられるもっとも一般的な依頼のひとつだ"
RESULTS:1 '= "盗賊の一団は古い山腹の遺跡を根城にしている。手配されている賊の頭を討ち取れば依頼達成となる"
RESULTS:3 '= "帝国が霧散し、大陸横断的な国家権力が消滅して以来、辺境地域は無法地帯と化している"
RESULTS:4 '= "自前の戦力を仕立てる余力のない者は冒険者や傭兵を頼ったが、"
RESULTS:5 '= "一方で食いっぱぐれた冒険者や傭兵が野盗に転向するケースも少なくはない"
RESULTS:6 '= "各地で諸勢力が勃興した今日においても、混沌とした辺境の情勢は昔とさほど変わらないようだ"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_12
RESULTS:0 '= @"難易度：@C:{カラー_注意}@★★@/C@"
RESULTS:1 '= "報酬：6,000ディナリ～,求人書類(初回のみ)"
RESULTS:2 '= "推奨パーティ人数:3人以上　　　推奨平均戦闘値：40以上"
RESULTS:3 '= "備考:定期性クエスト(4期毎)"
RESULTS:4 '= @"@C:{カラー_選択不可}@「冒険者も傭兵も盗賊もその日暮らしの流れ者である点には変わらん」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「コインの裏表みてぇなもんだし、コロコロ向きを変える奴もごまんといる」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「おまえさんはどうなんだろうな。まぁ俺の知ったことではないが」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_12(QUEST_ID)
#DIM QUEST_ID
;前提としてQUEST_0_ゴブリンの洞窟の攻略が必要
SIF !QST_SV_QUEST_FLAG:0
    RETURN 0
SIF QST_COOLTIME:QUEST_ID
    RETURN 0
RETURN 1

;-------------------------------------------------
;マップデータ定義
;クエストタイプが"GMAPの場合"マップ定義は外部マップデータに持っているので
; ERB側のマップデータにはEVENT_LISTさえ定義すればよい
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_12(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_12_野盗討伐"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "mapdata"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 1

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,1,]"
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,2,]"
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,4,100,101,102,]"
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,6,7,]"
_DIC_HO_MAPDATA:MAP_ID:6   '= "[EVENT_LIST:0,5,]"
_DIC_HO_MAPDATA:MAP_ID:7   '= "[EVENT_LIST:0,8,]"

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_12
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_12

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_12_ABORT
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;クエストの導入文
;-------------------------------------------------
@QST_INTRO_QUEST_12
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL

PRINTFORML %QST_LEADER_NAME()%は手配書の示す盗賊団の拠点に足を向けた
PRINTFORMW どうやら連中は古い時代の遺跡をねぐらにしているようだ
PRINTL
PRINTFORML 敵の戦力はさほど大きくはないようだが油断は禁物だ
PRINTFORMW 思わぬ罠が待ち受けている可能性もある
PRINTL
PRINTFORMW 常に警戒を怠るべきではないだろう……

FLAG:クエスト_中断を許可 = 0
;-------------------------------------------------
;イベントリストに登録するイベント情報
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 1
        CALL HO_PRINT_LOG("上古の遺跡のようだ。現代とも古帝国とも異なる、異質な様式の空間が広がっている", 1)
    CASE 2
        CALL HO_PRINT_LOG("部屋の中は青臭い、むせ返るような精臭と汗の臭いで満たされている", 1)
    CASE 3
        CALL HO_PRINT_LOG("盗賊が近隣から略奪した品々が部屋の隅に乱雑に積み上げられている", 1)
    CASE 5
        CALL HO_PRINT_LOG("天井には無数の配管が渡されており、壁には奇怪な装置の数々が並んでいる", 1)
    CASE 6
        CALL HO_PRINT_LOG("武器庫のようだ。粗野な作りの武器が乱雑に立て掛けられている", 1)
ENDSELECT

;-------------------------------------------------
;EV1
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_1
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_1
#DIM EVENT_ID = 1
PRINTFORMW 日が暮れるを待ち%QST_LEADER_NAME()%は野盗が拠点にしている山腹の遺跡に忍び込んだ
PRINTFORMW 盗賊団はさほど規模も大きくないのか警戒は手薄なようで、内部への侵入は難なく成功した
PRINT
PRINTFORML エントランスからは南北に二本の通路が伸びている
PRINTFORMW さて、どちらへ進もうか

;-------------------------------------------------
;EV2
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_2
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_2
#DIM CONST EVENT_ID = 1
#DIM CONST 派生イベントID = 3
#DIM EVENT_FLAG
#DIM CHARA_ID
CALL QST_GET_EVENT_FLAG(派生イベントID)
EVENT_FLAG = RESULT
;このイベントは後続の派生イベント(EV2)の選択結果で2回目以降の処理が変わるのでセットされたフラグを見て内容を分岐
SELECTCASE EVENT_FLAG
        ;初回
    CASE 0
        PRINTFORMW 小部屋の奥からくぐもった嬌声と水音が聞こえる……
        PRINTFORMW 陰から様子を伺ってみると、どうやら野盗どもが"戦利品"を弄んでいるようだ
        ;調教モードをクエスト -> 慰安に切替
        FLAG:ウフフフラグ = 1
        FLAG:調教モード = 調教_慰安
        TFLAG:調教ターン数 = 8
        FLAG:調教スキップ可 = 1

        CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
        ;竿役生成
        FOR LOCAL, 0, 3
            CALL QST_QUEST_12_CREATE_RAPER(0)
            CHARA_ID = RESULT
            CFLAG:CHARA_ID:調教参加フラグ = 1
            CFLAG:CHARA_ID:慰安参加者 = 1
            CFLAG:CHARA_ID:強制友好化 = 1
            CFLAG:CHARA_ID:慰安参加者非対象化 = 1
        NEXT
        ;適当に被害者を追加
        FOR LOCAL, 0, 4
            CALL QST_QUEST_12_CREATE_TRAIN_VICTIMS
        NEXT
        ;調教後に呼び出される後続イベント
        CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
    CASE 1
        PRINTFORMW 小部屋には狼藉者どもの死体が散乱している
        ;被害者を見捨てる
    CASE 2
        PRINTFORMW 野盗どもは欲望の赴くままに陵辱を続けている……
        PRINTFORMW 女の肢体を弄ぶのに夢中なのか、潜んでいる%QST_LEADER_NAME()%に気付く様子はない
ENDSELECT

;-------------------------------------------------
;EV3
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_3
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_3
#DIM CONST EVENT_ID = 3
#DIM EVENT_FLAG
PRINTFORMW 野盗どもは女を凌辱するのに夢中になっており、%QST_LEADER_NAME()%に気付く様子はない……
PRINTFORMW 彼女らを救出すべきだろうか……？
CALL COLOR_PRINTW("ここで騒ぎを起こせば敵の注意を引くかもしれない……", カラー_注意)
CALL ASK_YN("救出を試みる", "見捨てる")
IF RESULT == 0
    PRINTFORMW %QST_LEADER_NAME()%は隙だらけの狼藉者どもを強襲した！
    ;戦闘イベント
    CALL CC_INIT()
    FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
        CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
    NEXT
    CALL CC_ADDMOB("DEFENDER", "ならず者１", 2000, 2000, 30)
    CALL CC_ADDMOB("DEFENDER", "ならず者２", 2000, 2000, 30)
    CALL CC_ADDMOB("DEFENDER", "ならず者３", 2000, 2000, 30)
    CALL CLOSE_COMBAT()
    ;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
    IF RESULT == 1
        PRINTFORML 戦闘に勝利した%QST_LEADER_NAME()%は女性たちの解放に成功した
        PRINTFORMW 彼女らは涙ながらに礼を告げると、覚束ない足取りで出口へと向かった
        PRINTFORMW 謝礼として幾ばくかの金品を得た
        CALL COLOR_PRINTW("3,500ディナリを得た", カラー_注意)
        MONEY += 3500
        CALL COLOR_PRINTW(@"――　名誉点を15点得た！", カラー_注意)
        QST_PRESTIGE_POINT += 15
        CALL CC_ADD_COMBAT_EXP(4)
        CALL CC_APPLY_CHARA_HP()
        WAIT
        CALL QST_SET_EVENT_FLAG(EVENT_ID, 1)
    ELSE
        CALL CC_APPLY_CHARA_HP()
        CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%はならず者に返り討ちにされてしまった……", カラー_注意)
        PRINTFORMW 「ははっ、こいつぁは傑作だ。俺たちに勝てるとでも思ったのか？この間抜けめ」
        PRINTFORMW 「いまここでぶっ殺しちまってもいいが……まぁせっかくの新しい玩具だ、精々楽しませてもらうとしよう」
        PRINTFORMW 賊は敗北した%QST_LEADER_NAME()%を拘束すると、下卑た笑みを浮かべて弄びはじめた……
        CALL QST_QUEST_12_DEFEAT_EVENT
    ENDIF
ELSE
    PRINTFORMW 無用の危険を冒すべきではないと判断した%QST_LEADER_NAME()%はそっとその場を後にした……
    CALL QST_SET_EVENT_FLAG(EVENT_ID, 2)
ENDIF

;-------------------------------------------------
;EV4　倉庫
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_4
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_12_EVENT_4

PRINTFORML ここはどうやら山賊たちがかき集めた略奪品を収蔵している倉庫のようだ
PRINTFORMW なにか使えるものはないか物色してもよいし、そのまま奥へと進んでもよい
PRINTL
PRINTFORMW 奥へ進むなら、左か右か、どちらか選ぶ必要がある
PRINTL
PRINTFORMW 東側の小部屋は武器庫のようだ。中からは微かに人の気配がする……
PRINTL
PRINTFORML 東側の敵は強固に武装している可能性が高い
PRINTFORMW 先へ進むならまずは西側の部屋に忍び込み様子を見るのが賢明だと%ANAME(QST_RANDCHARA())%は思案した
PRINTL
PRINTFORMW ……さて、どちらに歩みを進めようか

CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV5 東の小部屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_5
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_5
#DIM EVENT_FLAG

CALL QST_GET_EVENT_FLAG(5)
IF !RESULT
    PRINTFORMW %QST_LEADER_NAME()%が部屋に入るや、中にいた賊は即座に武器を手に取り応戦の構えを取った
    PRINTFORML 元は傭兵か冒険者だったのだろう、何名かは明らかに場馴れしている様子だ
    PRINTFORMW 厄介な戦いになるかもしれない……
    ;戦闘イベント
    CALL CC_INIT()
    FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
        CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
    NEXT
    CALL CC_ADDMOB("DEFENDER", "傭兵くずれ１", 2500, 2500, 40, 0, 1)
    CALL CC_ADDMOB("DEFENDER", "冒険者くずれ１", 2000, 2000, 35)
    CALL CC_ADDMOB("DEFENDER", "ならず者１", 2000, 2000, 30)
    CALL CC_ADDMOB("DEFENDER", "ならず者２", 2000, 2000, 30)
    CALL CLOSE_COMBAT()
    ;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
    IF RESULT == 1
        CALL CC_ADD_COMBAT_EXP(6)
        CALL CC_APPLY_CHARA_HP()
        PRINTL
        PRINTFORMW 小部屋の敵を無力化することに成功した！
        CALL QST_SET_EVENT_FLAG(QST_NOWEV, 2)
    ELSE
        CALL CC_APPLY_CHARA_HP()
        CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%はならず者に返り討ちにされてしまった……", カラー_注意)
        PRINTFORMW 「ははっ、こいつぁは傑作だ。俺たちに勝てるとでも思ったのか？この間抜けめ」
        PRINTFORMW 「いまここでぶっ殺しちまってもいいが……まぁせっかくの新しい玩具だ、精々楽しませてもらうとしよう」
        PRINTFORMW 賊は敗北した%QST_LEADER_NAME()%を拘束すると、下卑た笑みを浮かべて弄びはじめた……
        CALL QST_QUEST_12_DEFEAT_EVENT
    ENDIF
ELSE
    PRINTFORMW 部屋の中は……がらんとしており、ひっくり返された椅子や乱雑にのけられた机から混乱の様子が伺える
    PRINTFORML 通風孔から送り込まれたガスの残り香か、微かな刺激臭が鼻孔をついた
    PRINTFORMW ……いずれにせよ脅威は排することができたようだ
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 2)
ENDIF

;-------------------------------------------------
;EV6 西の小部屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_6
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_12_EVENT_6

PRINTFORMW %QST_LEADER_NAME()%が入った部屋はことさら奇妙な空間であった
PRINTFORMW 天井に無数の配管が巡らされており、いくつかは壁際の装置類と接続されているのがみとめられる
PRINTFORMW これは何か気体を東側の部屋に流し込むための装置のようだ
PRINTL
PRINTFORMW 装置には実にわかりやすく粗野なドクロの模様が刻まれている。あとからならず者どもが刻んだのだろう
PRINTFORMW この装置が何か碌でもない作用を引き起こすものであることは間違いない
PRINTL
PRINTFORMW 配管に装置類、観察すればするほど部屋の異質さが際立ってくる
PRINTFORMW 今でこそ野盗たちの生活空間になっているものの、かつてはなにかを研究をするための施設だったのかもしれない
PRINTL
PRINTFORMW それが魔術なのか錬金術なのか、あるいはそのどちらでもない異質な技術なのか……調査の準備のない今は判断をつけられそうにない
PRINTL
PRINTFORMW さて、装置を稼働させれば東側の武器庫に何か影響を与えることができそうだ
;すでに東側の武器庫が排除されている
CALL QST_GET_EVENT_FLAG(5)
IF RESULT == 2
    PRINTFORMW ……が、武器庫の敵を殲滅した今、あえてこの不可解な装置に手を触れる理由もないだろう……
ELSE
    PRINTFORMW %QST_LEADER_NAME()%は古代の装置の稼働を試みても良いし、この怪しげな装置を見なかったことにしても良い
    PRINTL
    PRINTFORMW どうするにせよ、奥の大部屋に進む前に東の部屋、武器庫の小集団をまず掃除するのが賢明だ
    SIF QST_PARTYNUM > 1
        PRINTFORMW ……そう訴える%ANAME(QST_RANDCHARA())%の言葉に仲間は頷く
    PRINTL
    PRINTFORMW さもなければ大部屋での戦闘は一度に大勢の武装集団を相手取ることになるだろう……
ENDIF
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV7 奇妙な装置
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_7
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "奇妙な装置"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_12_EVENT_7

PRINTFORMW 装置は多数のツマミやレバーで構成されている
;すでに東側の武器庫が排除されている
CALL QST_GET_EVENT_FLAG(5)
IF RESULT == 2
    PRINTFORMW 今はあえてこの不可解な装置に手を触れる理由もないだろう……
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
    RETURN
ENDIF
PRINTFORMW さて……どうしたものだろうか？
CALL QST_ROLL_MULTI("知識,観察,身体,","解析を試みる(知識),よく観察してみる(観察),とりあえず叩いてみる(身体),", "50,100,150,", 7)
IF RESULT == 1
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW ……%ANAME(RESULT:1)%の見識によるとコモンウェルスの産業機械と類似点があり、操作できるかもしれないとのことだ
            PRINTFORMW そのような装置が古帝国以前に存在したという事実自体驚嘆すべきことだが、今は脅威の排除に集中する必要がある
            PRINTL
            PRINTFORMW %ANAME(RESULT:1)%が装置を操作すると機械類がゴトゴトと唸りを上げて稼働を始める
        CASE 1
            PRINTFORMW ……%ANAME(RESULT:1)%は装置と配管をじっくりと観察し、操作盤と各機械がどのように対応するのかアタリをつけていく
            PRINTFORMW そして直感を交えつつ装置を操作した
            PRINTL
            PRINTFORMW どうやら成功したようだ！%ANAME(RESULT:1)%が装置を操作すると機械類がゴトゴトと唸りを上げて稼働を始める
        CASE 2
            PRINTFORMW ……%ANAME(RESULT:1)%はお手上げと言わんばかりに肩をすくめると、装置を思いきりどついた！
            PRINTFORMW 少しの間をおいて機械類がゴトゴトと唸りを上げて稼働を始める。なんと成功してしまったようだ……
    ENDSELECT
    PRINTL
    PRINTFORMW 送風用と思われる回転翼が稼働を始め、配管を通して何かが隣室に送り込まれる
    PRINTFORMW やがてシュウシュウと噴出音が響き始めた
    PRINTL
    PRINTFORMW 「がぁぁぁぁ、目がぁ！目が痛ぇ！ああクソッ、何も見えねぇ！」
    PRINTFORMW 「馬鹿どもが……っ！機械を動かしやがったな！あれほど触るなって言い聞かせていたのによぉ！」
    PRINTL
    PRINTFORML 隣の部屋からはならず者たちが上げる絶叫が聞こえてくる
    PRINTFORMW 次いで乱雑に何かをひっくり返す音や破壊音、半狂乱にバタバタと部屋を飛び出していく足音が響く
    PRINTL
    PRINTFORMW ……どうやら労せず東の小部屋の脅威を排除することに成功したようだ！
    CALL QST_SET_EVENT_FLAG(5, 1)
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ELSEIF !RESULT
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW ……%ANAME(RESULT:1)%は装置を前にして首を捻っている。何もわかりそうにない……
        CASE 1
            PRINTFORMW ……%ANAME(RESULT:1)%は装置や配管を眺めてみたものの、結局時間だけが虚しく過ぎていった……
        CASE 2
            PRINTFORMW ……%ANAME(RESULT:1)%はやけくそ気味に装置をどついてみた。が、身体を襲う鈍い痛みの他は何も得られなかった……
    ENDSELECT
ELSEIF RESULT == -1
    PRINTFORMW ……得体の知れぬ装置をむやみに動かすべきではない、放っておこう
ENDIF
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;-------------------------------------------------
;EV8 北の大部屋
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_8
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_12_EVENT_8
#DIM EVENT_FLAG

PRINTFORMW %QST_LEADER_NAME()%が大部屋に押し入ると、室内にいた賊どもは一瞬虚ろを突かれたようだが、即座に応戦の構えを取った
PRINTL
PRINTFORMW 「チッ、ギルドの差し金か……！存外早かったな。おい野郎ども、ぐだぐだしてねぇでさっさと武器を取れ！」
PRINTFORMW 賊の頭領が檄をとばすや、斧や剣で武装した悪漢が%QST_LEADER_NAME()%の前に立ちはだかる
PRINTL
CALL CC_INIT()
CALL CC_ADDMOB("DEFENDER", "賊の頭領", 3000, 3000, 45, 0, 1)
CALL CC_ADDMOB("DEFENDER", "傭兵くずれ１", 2500, 2500, 40, 0, 1)
CALL CC_ADDMOB("DEFENDER", "ならず者１", 2000, 2000, 30)
CALL QST_GET_EVENT_FLAG(5)
;東の小部屋のならず者が排除されていない
IF !RESULT
    PRINTL
    PRINTFORMW さらに南側の入口から数名の賊が駆けつけて来た！
    PRINTFORMW 「お頭ぁ！なんの騒ぎで！？」
    PRINTFORMW 「……見りゃあわかるだろ、見りゃあよ。襲撃だ、襲撃。ボサッとしてねぇでとっとと加勢しやがれ！」
    PRINTL
    CALL CC_ADDMOB("DEFENDER", "傭兵くずれ２", 2500, 2500, 40, 0, 1)
    CALL CC_ADDMOB("DEFENDER", "冒険者くずれ１", 2000, 2000, 35)
    CALL CC_ADDMOB("DEFENDER", "ならず者２", 2000, 2000, 30)
ENDIF
PRINTFORMW ―――― 依頼を完遂するには連中をことごとく討滅する必要がある！
;戦闘イベント
FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
    CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
NEXT
CALL CLOSE_COMBAT()
;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
IF RESULT == 1
    CALL CC_ADD_COMBAT_EXP(7)
    CALL CC_APPLY_CHARA_HP()
    PRINTL
    PRINTFORMW %ANAME(QST_RANDCHARA())%は賊の頭目の攻撃をかわし、すかさず痛烈な一閃を見舞う
    PRINTFORMW 痛恨の一撃を受けた頭領はもんどりを打ってその場に崩れ落ちた
    PRINTL
    PRINTFORMW 頭目を失った残党は互いに顔を見合わせると、蜘蛛の子を散らすように逃げ出していった……
    PRINTL
    PRINTL
    PRINTFORMW 依頼は完了だ！討伐したことを示す証拠となり得そうなものを回収した%QST_LEADER_NAME()%はこの場を立ち去った
    PRINTL
    ;クリア報酬
    CALL COLOR_PRINTW(@"――　ギルドから成功報酬として6,000ディナリが支払われた", カラー_注意)
    MONEY += 6000
    ;初回攻略時に求人書類を付与
    IF QST_SV_QUEST_FLAG:QST_有効クエストID == 0
        CALL COLOR_PRINTW(@"――　求人書類を1つ手に入れた！", カラー_注意)
        HO_ITEM_STOCK:ITEM_求人書類 += 1
    ENDIF
    CALL COLOR_PRINTW(@"――　名誉点を20点得た！", カラー_注意)
    QST_PRESTIGE_POINT += 20

    QST_COOLTIME:QST_有効クエストID += 4
    QST_SV_QUEST_FLAG:QST_有効クエストID = 1

    ;リュミエール口上デイリーフラグ
    IF !K1_DFLAG_メイドの嗜み && LIST_INDEXOF(_LIST_QST_PARTY, TOSTR(NAME_TO_CHARA("リュミエール"))) != -1
        K1_DFLAG_メイドの嗜み = 1
    ENDIF
    TFLAG:クエスト_完了フラグ = 1
    CALL TRAIN_EXIT()
ELSE
    CALL CC_APPLY_CHARA_HP()
    CALL COLOR_PRINTW(@"%QST_LEADER_NAME()%はならず者に返り討ちにされてしまった……", カラー_注意)
    PRINTFORMW 「ギルドの質も落ちたもんだな。てんで歯ごたえがねぇ」
    PRINTFORMW 「さっさとぶっ殺しちまってもいいが……部下どもの"はけ口"には丁度いいだろう」
    PRINTFORMW 賊は敗北した%QST_LEADER_NAME()%を拘束すると、下卑た笑みを浮かべて弄びはじめた……
    CALL QST_QUEST_12_DEFEAT_EVENT
ENDIF
;-------------------------------------------------
;EV100 宝箱(技巧の宝玉)
;-------------------------------------------------
;イベント登録
@QST_ADD_EVENT_QUEST_12_EVENT_100
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"木箱"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_12_EVENT_100
#DIMS DYNAMIC _ITEM_LIST
#DIM  DYNAMIC CHEST_TYPE

_ITEM_LIST '= @"{ITEM_技巧の宝玉},1,"
CALL QST_CEV_CHEST(_ITEM_LIST)
IF RESULT
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;-------------------------------------------------
;EV101 宝箱(ポーション)
;-------------------------------------------------
;イベント登録
@QST_ADD_EVENT_QUEST_12_EVENT_101
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"施錠された木箱"
;取得済みの場合は非アクティブに
CALL QST_GET_EVENT_FLAG(QST_NOWEV)
DEBUGPRINTFORML RESULT:{RESULT}
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_12_EVENT_101
#DIMS DYNAMIC _ITEM_LIST
_ITEM_LIST '= @"{ITEM_ポーション},2,"
CALL QST_CEV_LOCKED_CHEST(_ITEM_LIST)
IF RESULT
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;-------------------------------------------------
;EV102 宝箱(衣服)
;-------------------------------------------------
;イベント登録
@QST_ADD_EVENT_QUEST_12_EVENT_102
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"木箱"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

;イベント実行内容
@QST_QUEST_12_EVENT_102
PRINTFORMW 木箱の中から略奪品と思われる衣類を見つけた
CALL CHANGE_CLOTH_STOCK(衣装部位_上下服, 15, 1)
CALL CHANGE_CLOTH_STOCK(衣装部位_上下服, 10, 1)
CALL CHANGE_CLOTH_STOCK(衣装部位_上下下着, 7, 1)
CALL PRINT_NOTICE("衣服:ぼろぼろの布切れを手に入れた")
CALL PRINT_NOTICE("衣服:ディアンドルを手に入れた")
CALL PRINT_NOTICE("衣服:ネグリジェを手に入れた")
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
CALL QST_DEACTIVATE_EVENT(QST_NOWEV)

;-------------------------------------------------
;EV10
;敗北イベント後の共通エンディング（クエスト失敗）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_10
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_10
#DIM CONST 派生イベントID = 11
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC HIGHEST_SEDUCTION_LV

PRINTL
PRINTFORMW 「あ？これで終わりだとでも思ったのか？」
PRINTFORMW 賊の頭領は激しい輪姦を受け息絶え絶えになる%QST_LEADER_NAME()%を引きずり上げると、首枷をはめ、身体の自由を奪う……
PRINTL
PRINTFORMW 「俺たちもなぁ……こんな狭苦しいところに野郎だけで押し込められて、溜まってんだよ」
PRINTFORMW 「これからは肉便器として散々使ってやるよ。感謝しな、命までは取らねぇおいてやる。今はな」
PRINTL
PRINTFORMW 「もっとも……部下どもは荒っぽくてなぁ、もしかしたら輪姦されてるうちに死んじまうかもしれん」
PRINTFORMW 「まぁそんときは自分の力不足を呪ってくれや、はははっ」
PRINTL
PRINTFORMW 「さーて……んじゃ、始めるか。お前ら、戦いの褒美だ。こいつらは好きにしていいぞ！」
PRINTL
PRINTFORMW 「さっすがお頭ぁ！話がわかりやすぜ」
PRINTFORMW 色めき立った賊たちが疲弊する%QST_LEADER_NAME()%の肢体に群がる……

;パーティの女性メンバーの誘蛾の色香LVの最高値を算出
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    IF IS_FEMALE(CHARA_ID)
        CALL SKILL_GET_LEVEL(CHARA_ID, スキル_大分類_性技, "SEDUCTION")
        SIF RESULT > HIGHEST_SEDUCTION_LV
            HIGHEST_SEDUCTION_LV = RESULT
    ENDIF
NEXT

CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
;パーティを追加
FOR LOCAL, 0, QST_PARTYNUM
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):凌辱被害者フラグ = 1
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):調教参加フラグ = 1
NEXT

;竿役生成
FOR LOCAL, 0, 4
    CALL QST_QUEST_12_CREATE_RAPER(0)
    CHARA_ID = RESULT
    CFLAG:CHARA_ID:調教参加フラグ = 1
    CFLAG:CHARA_ID:慰安参加者 = 1
    CFLAG:CHARA_ID:強制友好化 = 1
    CFLAG:CHARA_ID:慰安参加者非対象化 = 1
    IF HIGHEST_SEDUCTION_LV > 0
        MAXBASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力
        MAXBASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力
        ABL:CHARA_ID:倒錯度 =  ABL:CHARA_ID:倒錯度 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性技 = ABL:CHARA_ID:性技 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:欲望 = ABL:CHARA_ID:欲望 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性交 = ABL:CHARA_ID:性交 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:射精 = ABL:CHARA_ID:射精 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:Ｃ感 = ((ABL:CHARA_ID:射精 + ABL:CHARA_ID:欲望))
        IF RAND(2)
            TALENT:CHARA_ID:Ｃ敏感 = 1
            TALENT:CHARA_ID:絶倫 = 1
        ELSE
        ENDIF
    ENDIF
NEXT
;適当に被害者を追加
FOR LOCAL, 0, 2
    CALL QST_QUEST_12_CREATE_TRAIN_VICTIMS
NEXT
FOR LOCAL, 0, CHARANUM
    IF CFLAG:LOCAL:慰安参加者
        ;立ちバック系のみ許可
        VARSET COM_FORBIDDEN:LOCAL:0, 1
        COM_FORBIDDEN:LOCAL:T_N("服を破る") = 0
        COM_FORBIDDEN:LOCAL:T_N("二本挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("立ちバック") = 0
        COM_FORBIDDEN:LOCAL:T_N("順番に挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("代わる代わる挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("Ａ立ちバック") = 0
        COM_FORBIDDEN:LOCAL:T_N("順番にＡ挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("Ａ二本挿入") = 0
        COM_FORBIDDEN:LOCAL:T_N("イラマチオ") = 0
    ENDIF
NEXT

FLAG:ウフフフラグ = 1
FLAG:調教モード = 調教_慰安
IF HIGHEST_SEDUCTION_LV > 0
    TFLAG:調教ターン数 = 20 +  LIMIT(HIGHEST_SEDUCTION_LV ,1 , 3) * 4
ELSE
    TFLAG:調教ターン数 = 20
ENDIF

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
RETURN

;-------------------------------------------------
;EV11
;敗北イベント後の共通エンディング（クエスト失敗）
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_12_EVENT_11
#DIMS EVENT_TRIGGER = "TRAIN_EXIT"
RESULTS = %EVENT_TRIGGER%

@QST_QUEST_12_EVENT_11
#DIM CONST EVENT_ID = 5
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL
PRINTFORMW ……一体どれほど犯されたのだろうか？
PRINTFORMW %QST_LEADER_NAME()%はならず者どもに散々に辱められてしまった……
PRINTFORML 虜囚として散々な辱めを受けた後、幸運にもスキを突いて脱出する機会を得た%QST_LEADER_NAME()%は
PRINTFORMW 体液にまみれ、ぐったりと重い身体からなけなしの力を振り絞り、脱走を図った
PRINTL
PRINTFORMW なんとか近隣の町まで戻ることに成功した……
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
@QST_QUEST_12_DEFEAT_EVENT
#DIM CONST 派生イベントID = 10
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC HIGHEST_SEDUCTION_LV

FLAG:ウフフフラグ = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 12

;パーティの女性メンバーの誘蛾の色香LVの最高値を算出
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    IF IS_FEMALE(CHARA_ID)
        CALL SKILL_GET_LEVEL(CHARA_ID, スキル_大分類_性技, "SEDUCTION")
        SIF RESULT > HIGHEST_SEDUCTION_LV
            HIGHEST_SEDUCTION_LV = RESULT
    ENDIF
NEXT

CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, QST_PARTYNUM
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):凌辱被害者フラグ = 1
    CFLAG:(TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))):調教参加フラグ = 1
NEXT
;適当に被害者を追加
FOR LOCAL, 0, 2
    CALL QST_QUEST_12_CREATE_TRAIN_VICTIMS
NEXT
;竿役生成
FOR LOCAL:0, 0, 3
    CALL QST_QUEST_12_CREATE_RAPER(0)
    CHARA_ID = RESULT
    CFLAG:CHARA_ID:調教参加フラグ = 1
    CFLAG:CHARA_ID:慰安参加者 = 1
    CFLAG:CHARA_ID:強制友好化 = 1
    CFLAG:CHARA_ID:慰安参加者非対象化 = 1
    IF HIGHEST_SEDUCTION_LV > 0
        MAXBASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:体力 = MAXBASE:CHARA_ID:体力
        MAXBASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        BASE:CHARA_ID:気力 = MAXBASE:CHARA_ID:気力
        ABL:CHARA_ID:倒錯度 =  ABL:CHARA_ID:倒錯度 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性技 = ABL:CHARA_ID:性技 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:欲望 = ABL:CHARA_ID:欲望 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:性交 = ABL:CHARA_ID:性交 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:射精 = ABL:CHARA_ID:射精 * LIMIT(HIGHEST_SEDUCTION_LV, 1, 3)
        ABL:CHARA_ID:Ｃ感 = ((ABL:CHARA_ID:射精 + ABL:CHARA_ID:欲望))
        IF RAND(2)
            TALENT:CHARA_ID:Ｃ敏感 = 1
            TALENT:CHARA_ID:絶倫 = 1
        ELSE
        ENDIF
    ENDIF
NEXT

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)
RETURN

;-------------------------------------------------
;慰安モブの定義
;-------------------------------------------------
@QST_QUEST_12_CREATE_TRAIN_VICTIMS()
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %SPLIT_RAND("冒険者/村娘/行商人/性奴隷",1)%

CALL CREATE_RANDOM_CHARA(0, -1, 1)
FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
CALLNAME:対象 = %名前%
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = 1
ABL:対象:性知識 =  3
ABL:対象:Ｃ感 = 28
ABL:対象:奉仕 = 1
ABL:対象:性技 = RAND(10, 14)
ABL:対象:欲望 = RAND(3, 7)
ABL:対象:性交 = RAND(13, 15)
ABL:対象:マゾ = RAND(0, 3)
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = RAND(0, 300)
CFLAG:対象:調教参加フラグ  = 1
CFLAG:対象:主導除外  = 1
CFLAG:対象:慰安モブ = 1
CFLAG:対象:凌辱被害者フラグ = 1

;-------------------------------------------------
;慰安モブの定義
;-------------------------------------------------
@QST_QUEST_12_CREATE_RAPER(SEX)
#DIM SEX
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %SPLIT_RAND("ならず者/悪漢/野盗/傭兵くずれ/盗賊/冒険者くずれ",1)%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)
FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
CALLNAME:対象 = %名前%
TALENT:対象:処女 = 0
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
ABL:対象:戦闘 =  RAND(35, 50)
ABL:対象:性知識 = 3
ABL:対象:奉仕 = 1
ABL:対象:性技 = RAND(0, 4)
ABL:対象:欲望 = RAND(3, 7)
ABL:対象:性交 = RAND(3, 5)
ABL:対象:射精 = RAND(3, 5)
ABL:対象:サド = RAND(4, 8)
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = RAND(0, 300)
CFLAG:対象:調教参加フラグ = 1
CFLAG:対象:慰安モブ = 1

COM_TENDENCY:対象:GETNUM(TRAINNAME, "服を破る") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "代わる代わる挿入") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "代わる代わるＡ挿入") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "順番に挿入") = 3
COM_TENDENCY:対象:GETNUM(TRAINNAME, "順番にＡ挿入") = 3

LOCALS = %S_R("male01/male02/male03/shota01/shota02")%
CSTR:対象:顔グラファイル名 '= @"%LOCALS%.webp"
CSTR:対象:キャラ顔グラ '= @"%LOCALS%.webp"
CFLAG:対象:ポートレイト画像タイプ = 2
CFLAG:対象:汎用口上パターン = 2002
CFLAG:対象:口上カラーコード = カラー_選択不可

RETURN 対象


code:
MAP_NAME = "盗賊の拠点"
MAP_IMAGE = "dg00.webp"
PATH_COLOR = "0xff00"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0x2c2c2c"


code:
<DocumentElement>
  <GMAPDATA>
    <id>638135971342404270</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>エントランス</NODE_NAME>
    <POS_X>160</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971426815879</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>0,2,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971448656173</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>140</POS_X>
    <POS_Y>300</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971479074879</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>280</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>1,5,6,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971642223563</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>140</POS_Y>
    <PATH_LIST>6,7,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971679283709</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>小部屋</NODE_NAME>
    <POS_X>420</POS_X>
    <POS_Y>180</POS_Y>
    <PATH_LIST>5,7,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638135971722553106</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>大部屋</NODE_NAME>
    <POS_X>480</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>6,5,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_13:辺境の教会
;-------------------------------------------------
@QST_TITLE_QUEST_13
RESULTS '= "辺境の教会"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_13
RESULTS:0 '= @"教会は来るものを拒まない。もし%ANAME(MASTER)%が望むのならば足を運んでもよいだろう"
RESULTS:1 '= "望むなら祈りを捧げたり、後ろめたい罪があるあらば告白しても良い"
RESULTS:4 '= "古帝国が正教を定めて以来、中央大陸各地に教会が建築され土着の信仰に取って代わった"
RESULTS:5 '= "帝国亡き後も正教は大陸の人々に広く信仰され、教会は地域の信仰の拠点となっている"
RESULTS:6 '= "ただし帝国の崩壊によって中央による指導が行き届かなくなると土着の宗教と教義が入り混じったり"
RESULTS:7 '= "異端化・カルト化し変質するケースも少なくはないようだ"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_13
RESULTS:0 '= @"難易度：@C:{カラー_注意}@命の危険はおそらくない@/C@"
RESULTS:1 '= "報酬：-"
RESULTS:2 '= "推奨パーティ人数:1人以上"
RESULTS:3 '= ""
RESULTS:4 '= @"@C:{カラー_選択不可}@「なんだお前さんそんなに信心深かったのか？」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「祈ってる手が空いているなら剣でも握ってればいいだろうに」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_13(QUEST_ID)
#DIM QUEST_ID
;前提としてQUEST_0_ゴブリンの洞窟の攻略が必要
SIF !QST_SV_QUEST_FLAG:0
    RETURN 0
SIF QST_COOLTIME:QUEST_ID
    RETURN 0
RETURN 1

;-------------------------------------------------
;マップデータ定義
;クエストタイプが"GMAPの場合"マップ定義は外部マップデータに持っているので
; ERB側のマップデータにはEVENT_LISTさえ定義すればよい
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_13(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_13_辺境の教会"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "辺境の教会"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 1

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,2,3,4,]"
_DIC_HO_MAPDATA:MAP_ID:4   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,50,51,59,]"

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_13
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_13

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_13_ABORT
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;クエストの導入文
;-------------------------------------------------
@QST_INTRO_QUEST_13
; PRINTL
; PRINTFORMW ……………………
; PRINTFORMW ………………
; PRINTFORMW …………
; PRINTL

; PRINTFORML %QST_LEADER_NAME()%は領地の外れにある教会に足を運んだ
; PRINTFORMW 最低限の礼拝堂を備えた小さな教会で、人の気配はほぼ無いに等しい
; PRINTL
; PRINTFORMW 村人の話によると今は若い姉妹のシスターが二人で切り盛りしているとのことだ

;ゲストユニークキャラ、ロザリィを追加
LOCAL = NAME_TO_CHARA("シスター・ロザリィ")
IF LOCAL == -1
    CALL QST_QUEST_13_CREATE_UNIQUE_MOB1()
    LOCAL = RESULT
ENDIF
CALL QST_SET_FLAGMAP("NPC_ID", TOSTR(LOCAL))
;クエスト用フラグ管理変数にCHARA_IDを登録

;ゲストユニークキャラ、シャーリィを追加
LOCAL = NAME_TO_CHARA("シスター・シャーリィ")
IF LOCAL == -1
    CALL QST_QUEST_13_CREATE_UNIQUE_MOB2()
    LOCAL = RESULT
ENDIF
;クエスト用フラグ管理変数にCHARA_IDを登録
CALL QST_SET_FLAGMAP("NPC_ID2", TOSTR(LOCAL))

FLAG:クエスト_中断を許可 = 1

;-------------------------------------------------
;イベントリストに登録するイベント情報
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_13_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 1
        CALL HO_PRINT_LOG("ホールは静寂に包まれている。人の気配は無い", 1)
    CASE 2
        CALL HO_PRINT_LOG("部屋の一角に厳しい装飾の施された懺悔室がぽつんと置かれている", 1)
    CASE 4
        CALL HO_PRINT_LOG("薄暗い部屋を燭台の頼りない明かりが照らしている", 1)
    CASE 5
        CALL HO_PRINT_LOG("薄暗く、静寂に包まれた礼拝堂で僅かな信徒が祈りを捧げている", 1)
ENDSELECT

;-------------------------------------------------
;EV2
;ロザリィに告解
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_2
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "懺悔室で告解を始める"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_2
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC NPC_ID

; IF RAND:2
;     NPC_ID = NAME_TO_CHARA("シスター・ロザリィ")
; ELSE
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
; ENDIF
CALL QST_SET_FLAGMAP("NPC_ID", TOSTR(NPC_ID))
CALL QST_GET_EVENT_FLAG(QST_NOWEV)
IF !RESULT
    PRINTFORML 懺悔室がある。望むならシスター・%ANAME(NPC_ID)%に悩み事や罪を告白しても良い
    CALL QST_SELECT_CHARA(, "さて誰に告解させようか……")
    CHARA_ID = RESULT
    IF CHARA_ID == -1
        CALL KJ_PRINT(NPC_ID, "神はいつでも貴方の言葉に耳を傾けています。もし自身の歩む道に疑問が生じたなら、いつでもいらしてください", 1)
    ELSE
        CALL QST_SET_FLAGMAP("CHARA_ID", TOSTR(CHARA_ID))
        CALL QST_ACTIVATE_EVENT(3)
        CALL QST_ACTIVATE_EVENT(4)
        CALL QST_SET_EVENT_SLOT_NAME(, "懺悔室を出る")
        CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
        ;移動を完全に制限
        FLAG:移動制限フラグ = 2
        CALL KJ_PRINT(NPC_ID, "貴方の迷いをお聞きいたしましょう。さぁ、お入りになって", 1)
    ENDIF
ELSE
    CALL QST_DEACTIVATE_EVENT(3)
    CALL QST_DEACTIVATE_EVENT(4)
    CALL QST_SET_EVENT_SLOT_NAME(, "懺悔室で告解を始める")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    ;移動制限を解除
    FLAG:移動制限フラグ = 0
    CALL KJ_PRINT(NPC_ID, @"%ANAME(CHARA_ID)%、神はいつでも貴方の言葉に耳を傾けていますよ。もし自身の歩む道に疑問が生じたなら、いつでもいらしてください", 1)
ENDIF

;-------------------------------------------------
;EV3
;ロザリィに告解
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_3
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "告解する"

CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_3
#DIM NPC_ID
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))
CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MTAR(CHARA_ID)
;CALL QST_EXEC_DUMMYCOM(T_N("フェラチオ"))

;MASTERが対象ならソース処理は会話と同じものを使う
IF CHARA_ID == MASTER
    CALLFORM COM{T_N("会話")}
ELSE

ENDIF
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は懺悔室で%ANAME(NPC_ID)%に自身の悩みを打ち明けた
;擬似コマンド実行
CALL QST_EXEC_DUMMYCOM()

;-------------------------------------------------
;EV4
;ロザリィに告解フェラ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_4
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = @"@C:{カラー_ピンク}@❤告解する❤@/C@"

CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_4
#DIM EVENT_FLAG
#DIM NPC_ID
#DIM NPC_ID2
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
NPC_ID2 = TOINT(QST_GET_FLAGMAP("NPC_ID2"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))

CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
CFLAG:NPC_ID:スケベモブ = 1
IF NPC_ID2
    CFLAG:NPC_ID2:調教参加フラグ = 1
    CFLAG:NPC_ID2:強制友好化 = 1
    CFLAG:NPC_ID2:スケベモブ = 1
ENDIF
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MPLY(NPC_ID2)
CALL ADD_MTAR(CHARA_ID)

;ソース処理は会話と同じものを使う
CALLFORM COM{T_N("会話")}
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は懺悔室で%ANAME(NPC_ID)%に自身の"悩み"を処理させた……
IF HAS_PENIS(CHARA_ID)
    ;擬似コマンド実行
    CALL QST_EXEC_DUMMYCOM(T_N("フェラチオ"))
ELSE
    CALL QST_EXEC_DUMMYCOM(T_N("クンニ"))
ENDIF

;ダミーコマンド実行後に呼び出されるイベント
@QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT
#DIM NPC_ID
#DIM NPC_ID2
#DIM 射精量
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
NPC_ID2 = TOINT(QST_GET_FLAGMAP("NPC_ID2"))

射精量  = HO_CHECK_CUM_AMOUNT(NPC_ID, 0)
射精量  += HO_CHECK_CUM_AMOUNT(NPC_ID2, 0)
IF 射精量 >= 200
    CALL QST_SET_ROOM_EFFECT("むせ返るような精臭", "0")
ENDIF
CFLAG:NPC_ID:調教参加フラグ = 0
CFLAG:NPC_ID:強制友好化 = 0
CFLAG:NPC_ID:スケベモブ = 0
IF NPC_ID2
    CFLAG:NPC_ID2:調教参加フラグ = 0
    CFLAG:NPC_ID2:強制友好化 = 0
    CFLAG:NPC_ID2:スケベモブ = 0
ENDIF
FLAG:主導権所有者 = -1
CALL INIT_EQUIP()
CALL CLEAR_MPLY
CALL CLEAR_MTAR

IF GROUPMATCH(SELECTCOM, T_N("フェラチオ"), T_N("クンニ"))
    {
        LOCALS '=
            @"%S_R("密室の中で/厳かで窮屈な空間の中で/一枚の薄い壁の向こうで/壁を隔てた向こうで")%"
            +@"%S_R("ぴちゃぴちゃと/にちゃにちゃと/くちゅくちゅと")%肉を啜る、"
            +@"%S_R("淫猥な水音/淫らな音/色に満ちた音")%"
            +@"が響いている"
    }
    CALL COLOR_PRINTL(@"%LOCALS%", カラー_ピンク)
ENDIF
; #endregion

;-------------------------------------------------
;礼拝堂
;-------------------------------------------------
;-------------------------------------------------
;ロザリィと話す
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_50
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "ロザリィと話す"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_50
#DIM NPC_ID
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))
CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MTAR(CHARA_ID)

;MASTERが対象ならソース処理は会話と同じものを使う
IF CHARA_ID == MASTER
    CALLFORM COM{T_N("会話")}
ELSE
ENDIF
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は礼拝堂で%ANAME(NPC_ID)%と談笑した
;擬似コマンド実行
CALL QST_EXEC_DUMMYCOM()

;-------------------------------------------------
;シャーリィと話す
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_13_EVENT_51
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "シャーリィと話す"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_51
#DIM NPC_ID
#DIM CHARA_ID
NPC_ID = TOINT(QST_GET_FLAGMAP("NPC_ID2"))
CHARA_ID = TOINT(QST_GET_FLAGMAP("CHARA_ID"))
CFLAG:NPC_ID:調教参加フラグ = 1
CFLAG:NPC_ID:強制友好化 = 1
FLAG:主導権所有者 = NPC_ID
CALL CLEAR_MPLY
CALL CLEAR_MTAR
CALL ADD_MPLY(NPC_ID)
CALL ADD_MTAR(CHARA_ID)

;MASTERが対象ならソース処理は会話と同じものを使う
IF CHARA_ID == MASTER
    CALLFORM COM{T_N("会話")}
ELSE
ENDIF
;コマンド実行後に呼び出すイベントを登録
MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE_ONCE", "QST_QUEST_13_EVENT_CONFESSION_RELEASE_EVENT"

PRINTFORMW %ANAME(CHARA_ID)%は礼拝堂で%ANAME(NPC_ID)%と談笑した
;擬似コマンド実行
CALL QST_EXEC_DUMMYCOM()

@QST_ADD_EVENT_QUEST_13_EVENT_59
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "信徒席"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_13_EVENT_59
#DIM EVENT_FLAG
#DIM CHARA_ID

;擬似設備の使用
;MODULE_ID:5(机)を擬似的に使用している状態に
{
    CALL QST_CEV_USE_DUMMY_MODULE(
        @"[MODULE_ID:5]"
        +@"[PAUSE_BGM:1]"
        +@"[ALLOW_COM:1]"
        +@"[MODULE_NAME:信徒席]"
        +@"[USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は信徒席に座った……]"
        +@"[MODULE_NAME_AT_USING:席を立つ]"
        +@"[STOP_USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は席を立った]"
        +@"[REDRESS_AT_STOP:1]"
        )
}

;-------------------------------------------------
;クエスト用ユニークNPC
;-------------------------------------------------
@QST_QUEST_13_CREATE_UNIQUE_MOB1()
#DIM SEX = 1
#DIMS MOB_NAME = "ロザリィ"
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %MOB_NAME%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)

FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
;ユニークキャラとして一度だけ生成するのでCSTR:99を登録
CSTR:対象:99 '= "シスター・ロザリィ"
CALLNAME:対象 = %名前%
BASE:対象:体力 = 2500
BASE:対象:気力 = 2500
MAXBASE:対象:体力 = 2500
MAXBASE:対象:気力 = 2500
TALENT:対象:処女 = 1
TALENT:対象:キス未経験 = 0
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 0
TALENT:対象:絶倫 = 1

CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
ABL:対象:戦闘 =  45
ABL:対象:魔術 =  0
ABL:対象:性知識 = 5
ABL:対象:奉仕 = 6
ABL:対象:精愛 = 7
ABL:対象:性技 = 6
ABL:対象:欲望 = 7
ABL:対象:性交 = 4
ABL:対象:サド = 4
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = 500

CSTR:対象:120 '= "CC1_後髪_5"
CSTR:対象:121 '= "CC1_身体_0"
CSTR:対象:122 '= "CC1_顔_0"
CSTR:対象:123 '= "CC1_瞳_2"
CSTR:対象:124 '= "CC1_目_2"
CSTR:対象:128 '= "CC1_服_ローブ_0"
CSTR:対象:129 '= "CC1_前髪_4"
CSTR:対象:150 '= "384:344:288"
CSTR:対象:151 '= "白"
CSTR:対象:152 '= "白"
CSTR:対象:153 '= "510:100:204"
CSTR:対象:158 '= "256:248:348"
CSTR:対象:159 '= "384:344:288"

CFLAG:対象:汎用口上パターン = 2010
CFLAG:対象:口上カラーコード = カラー_クリーム

TALENT:対象:バストサイズ = 2
TALENT:対象:ヒップサイズ = 1
TALENT:対象:体格 = 0

CALL SET_BODYSIZE(対象)

RETURN 対象

@QST_QUEST_13_CREATE_UNIQUE_MOB2()
#DIM SEX = 1
#DIMS MOB_NAME = "シャーリィ"
#DIM 対象
#DIMS 名前
#DIM 主導除外フラグ

名前 = %MOB_NAME%

CALL CREATE_RANDOM_CHARA(0, -1, SEX)

FLAG:汎用武将カウント --
対象 = RESULT
IF LOCAL == -1
    RETURN -1
ENDIF

NO:対象 = GET_EMPTY_NO()
NAME:対象 = %名前%
;ユニークキャラとして一度だけ生成するのでCSTR:99を登録
CSTR:対象:99 '= "シスター・シャーリィ"
CALLNAME:対象 = %名前%
BASE:対象:体力 = 2500
BASE:対象:気力 = 2500
MAXBASE:対象:体力 = 2500
MAXBASE:対象:気力 = 2500
TALENT:対象:処女 = 1
TALENT:対象:キス未経験 = 1
TALENT:対象:童貞 = 0
TALENT:対象:アナル処女 = 1
TALENT:対象:絶倫 = 1
CSTR:対象:0 =
CSTR:対象:1 =
TALENT:対象:性別 = SEX
ABL:対象:戦闘 =  45
ABL:対象:魔術 =  0
ABL:対象:性知識 = 5
ABL:対象:奉仕 = 6
ABL:対象:精愛 = 0
ABL:対象:性技 = 0
ABL:対象:欲望 = 0
ABL:対象:性交 = 0
ABL:対象:サド = 0
ABL:対象:主導度Ｕ = RAND(0, 500)
ABL:対象:倒錯度 = 250

CSTR:対象:120 '= "CC1_後髪_11"
CSTR:対象:121 '= "CC1_身体_0"
CSTR:対象:122 '= "CC1_顔_0"
CSTR:対象:123 '= "CC1_瞳_2"
CSTR:対象:124 '= "CC1_目_2"
CSTR:対象:128 '= "CC1_服_ローブ_2"
CSTR:対象:129 '= "CC1_前髪_22"
CSTR:対象:130 '= "CC1_眉_0"
CSTR:対象:132 '= "CC1_帽子_ヴェール_2"
CSTR:対象:150 '= "黄"
CSTR:対象:151 '= "白"
CSTR:対象:152 '= "白"
CSTR:対象:153 '= "赤"
CSTR:対象:158 '= "白"
CSTR:対象:159 '= "黄"

CFLAG:対象:汎用口上パターン = 2010
CFLAG:対象:口上カラーコード = カラー_クリーム

TALENT:対象:バストサイズ = -1
TALENT:対象:ヒップサイズ = 0
TALENT:対象:体格 = -1

CALL SET_BODYSIZE(対象)

RETURN 対象


code:
MAP_NAME = "辺境の教会"
MAP_IMAGE = "dg08.webp"
PATH_COLOR = "0xff00"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0x4834FF"

path: ERB/QUEST/クエストデータ/QUEST_13_辺境の教会/辺境の教会.txt


code:
<DocumentElement>
  <GMAPDATA>
    <id>638137465039467278</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>入口</NODE_NAME>
    <POS_X>200</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465088867759</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>エントランスホール</NODE_NAME>
    <POS_X>260</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>0,4,2,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465137187875</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>礼拝堂-右翼</NODE_NAME>
    <POS_X>360</POS_X>
    <POS_Y>280</POS_Y>
    <PATH_LIST>1,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465168817767</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>懺悔室</NODE_NAME>
    <POS_X>440</POS_X>
    <POS_Y>280</POS_Y>
    <PATH_LIST>2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465199298193</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>礼拝堂-左翼</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>120</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638137465241837690</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>祭壇</NODE_NAME>
    <POS_X>440</POS_X>
    <POS_Y>100</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>

path: ERB/QUEST/クエストデータ/QUEST_13_辺境の教会/辺境の教会.xml


code:
;露店の商品在庫(商品在庫を管理・セーブする必要がない場合は、ERHへの定義は控えた方が良い)
#DIMS SAVEDATA QUEST_14_DIC_SHOP_STOCK_0, 10
#DIMS SAVEDATA QUEST_14_DIC_SHOP_STOCK_1, 10
#DIMS SAVEDATA QUEST_14_DIC_SHOP_STOCK_2, 10
;ダートトラックミニゲーム用の変数
#DIMS QUEST_14_C_Name


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_14:交易都市フォルメン
;-------------------------------------------------
@QST_TITLE_QUEST_14
RESULTS '= "交易都市フォルメン"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_14
RESULTS '= "周期,街,"

;クエスト選択画面で呼び出される概要説明
;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_14
RESULTS:0 '= "フォルメンは中央大陸東部の自由都市国家郡の一つである"
RESULTS:1 '= "白霜山脈の東に端を発し北海へ注ぐ大河と東西をつなぐ街道の交差点に位置し、商都として繁栄している"
RESULTS:2 '= "フォルメン市の歴史は深く、古帝国の成立以前より既に人間の居留地が存在していたとされる"
RESULTS:4 '= "街の西側を流れる大河は比較的大型の船舶も航行可能であり、大陸内陸部に位置するにもかかわらず"
RESULTS:5 '= "さながら貿易港のような賑わいを見せており、大陸の内外から運び込まれた多様な品々が市場を彩っている"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_14
RESULTS:0 '= @"難易度：- (実装途中のクエストです)"
RESULTS:1 '= "報酬：-"
RESULTS:2 '= "推奨パーティ人数:1人以上"
RESULTS:3 '= ""
RESULTS:4 '= @"@C:{カラー_選択不可}@「フォルメンの河辺で陽が沈むまで釣り竿を垂れて、その後に踊る子鹿亭で一杯やるんだ」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「あそこの魚料理をつまみながらこう、黒ビールをくいっとやるとそりゃもうそりゃもう絶品でなぁ……」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「おっと……まぁ気が向いたら行ってるみるといい」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_14(QUEST_ID)
#DIM QUEST_ID
;前提としてQUEST_0_ゴブリンの洞窟の攻略が必要
; SIF !QST_SV_QUEST_FLAG:0
;     RETURN 0
; SIF QST_COOLTIME:QUEST_ID
;     RETURN 0
RETURN 1

;-------------------------------------------------
;マップデータ定義
;クエストタイプが"GMAPの場合"マップ定義は外部マップデータに持っているので
; ERB側のマップデータにはEVENT_LISTさえ定義すればよい
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_14(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_14_交易都市フォルメン"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "交易都市フォルメン"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 1

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,]"     ;街道
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,10,11,]"     ;北市門
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,20,21,]"     ;広場
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,30,31,39,]"  ;教会
_DIC_HO_MAPDATA:MAP_ID:4   '= "[EVENT_LIST:0,40,41,42,43,44,45,]"   ;市場
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,]"     ;踊る子鹿亭前
_DIC_HO_MAPDATA:MAP_ID:6   '= "[EVENT_LIST:0,60,302,]"  ;ウォーターフロント
_DIC_HO_MAPDATA:MAP_ID:7   '= "[EVENT_LIST:0,]"     ;船着場
_DIC_HO_MAPDATA:MAP_ID:8   '= "[EVENT_LIST:0,]"     ;船着場
_DIC_HO_MAPDATA:MAP_ID:9   '= "[EVENT_LIST:0,]"     ;南大通り
_DIC_HO_MAPDATA:MAP_ID:10  '= "[EVENT_LIST:0,100,]"     ;南市門
_DIC_HO_MAPDATA:MAP_ID:11  '= "[EVENT_LIST:0,301,110,]"  ;スラム
_DIC_HO_MAPDATA:MAP_ID:12  '= "[EVENT_LIST:0,301,120,123,]"  ;スラム
_DIC_HO_MAPDATA:MAP_ID:13  '= "[EVENT_LIST:0,130,131,132,133,134,135,136,137,]"  ;踊る子鹿亭
_DIC_HO_MAPDATA:MAP_ID:14  '= "[EVENT_LIST:0,140,142,]"   ;場末の酒場『ヴァージン』
_DIC_HO_MAPDATA:MAP_ID:15  '= "[EVENT_LIST:0,150,151,152,]"   ;河辺のダートコース

;-------------------------------------------------
;クエストのタイプ
;-------------------------------------------------
@QST_TYPE_QUEST_14
RESULTS '= "GMAP"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_14

;-------------------------------------------------
;クエストの探索状態保持設定
;-------------------------------------------------
@QST_QUEST_14_SAVE_EXPLORED
RETURN 1

;-------------------------------------------------
; クエスト中止時の処理
; QST_QUEST_{QUEST_ID}_ABORT は撤退コマンド実行時に呼ばれる
;-------------------------------------------------
@QST_QUEST_14_ABORT
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;クエストの導入文
;-------------------------------------------------
@QST_INTRO_QUEST_14
CALL PRINT_TITLE("交易都市フォルメン", "クエスト")

PRINTFORML 大河に沿って街道沿いに南へ馬車を進め、丘を越えるとフォルメンの街が見えてきた
PRINTFORML 赤レンガの背の高い市壁が街の外周を囲っており、両側に背の高い塔をもった市門が入口に構えられている
PRINTL
PRINTFORML 門からは行商人や通行人が盛んに往来している様子が伺える
PRINTFORML 左手に目をやるとノルディアの快速帆船が穏やかなパラドゥイン河の流れを遡上して船着場に向かうのが見えた
PRINTFORMW 大河は数多くの商船が行き交っており、交通、貿易の大動脈として機能しているようだ

;ゲーム内時刻を8:00にセット
HO_TIME = HO_CLOCK(800)
FLAG:クエスト_中断を許可 = 1
;居住区で実行可能なコマンドの一部を実行可能にするフラグ
FLAG:クエスト_交流コマンド許可 = 1
;0時にクエスト強制終了
FLAG:QST_終了時刻 = HO_CLOCK(2400)
;BGMセット
CALL FL_PLAYBGM("Nostalgic_4.ogg")
;-------------------------------------------------
;イベントリストに登録するイベント情報
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;エリア入場時にログ出力する情景描写
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_0
; #region EV0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_14_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 0 ;街道
        CALL HO_PRINT_LOG("キャラバンがフォルメンに向かって長い隊列をつくっている", 1)
    CASE 1, 10 ;市門
        CALL HO_PRINT_LOG("馬車や通行人がひっきりなしに往来している", 1)
    CASE 2 ;広場
        CALL HO_PRINT_LOG("広場は通行人や商人が行き交っており、大道芸人が喧騒に華を添える", 1)
    CASE 3 ;教会
        CALL HO_PRINT_LOG("背の高い尖塔を持つ教会が街の中心にそびえ立っている", 1)
    CASE 4 ;市場
        CALL HO_PRINT_LOG("市場は買い物客や行商人でごった返している", 1)
    CASE 5 ;踊る子鹿亭前
        CALL HO_PRINT_LOG("酒場の入口には後脚で立ち、踊っている様子の鹿の看板がぶら下がっている", 1)
    CASE 6 ;ウォーターフロント
        CALL HO_PRINT_LOG("大河を行き交う大小様々な船舶が見える", 1)
    CASE 7, 8 ;船着場
        CALL HO_PRINT_LOG("船着場には多数の船舶が係留され、船員や商人が積荷を運んでいる", 1)
    CASE 9 ;南大通り
        CALL HO_PRINT_LOG("両脇に商店が連なる大通りは数多の人で賑わっている", 1)
    CASE 11, 12 ;スラム
        CALL HO_PRINT_LOG("裏通りへ入ると、うらぶれた雰囲気のスラムが広がっている", 1)
    CASE 13 ;踊る子鹿亭
        CALL HO_PRINT_LOG("店内は商人や冒険者、住人に観光客で大変な賑わいを見せている", 1)
ENDSELECT
; #endregion

;-------------------------------------------------
;EVXX イベント名
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_XX
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "雛形"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_14_EVENT_XX

;-------------------------------------------------
;EV10- 北市門
;-------------------------------------------------
;EV10 北市門の衛兵、会話イベント
@QST_ADD_EVENT_QUEST_14_EVENT_10
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "北市門の衛兵"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_10
PRINTFORML 「盗賊団だのなんだの近頃フロリアからの難民に紛れて厄介な連中が流れ込んでくることが増えた」
PRINTFORMW 「お前さん方は厄介事を増やしてくれるなよ？」

;EV10 北市門の衛兵、会話イベント
@QST_ADD_EVENT_QUEST_14_EVENT_11
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "難民の列"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_11
PRINTFORML …門から市街にかけての人々の流れを観察してみると、ぼろ布を纏ったひどくみずぼらしい人々の姿がしばしば見られた
PRINTFORML 彼らの性別、年齢構成は様々だが、その顔は皆一様に生気が無く、瞳からは光が失われている
PRINTFORMW 恐らくは王党派、共和派に分断され、終わりの見えなくなったフロリアの内戦から逃れた人々なのだろう
PRINTL
PRINTFORML フロリアの戦災難民達は、列を為して市街地中央の教会を目指しているようだ
PRINTFORMW 飢えや衰弱からの救いを教会の炊き出しに、そして疲弊した心の救いを神に求めているのであろう…

;-------------------------------------------------
;EV20- 広場
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_20
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "演台"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_14_EVENT_20
#DIM EVENT_FLAG
CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    PRINTFORML %QST_LEADER_NAME()%は演台に立った
    CALL QST_SET_EVENT_SLOT_NAME(, @"演台を降りる")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;擬似的に設備:演台(ID=901)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 901
    ;パフォーマンス関連の初期化処理
    CALL PERFORMANCE_INIT()
    TFLAG:公演_難易度 = 0
    WAIT
ELSE
    PRINTFORMW %QST_LEADER_NAME()%は演台から降りた
    CALL QST_SET_EVENT_SLOT_NAME(, @"演台")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    TFLAG:クエスト_ダミー設備ID = 0
    ;パフォーマンス関連の初期化処理
    CALL PERFORMANCE_INIT()
ENDIF

;-------------------------------------------------
;EV21 ベンチ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_21
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "ベンチ"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_14_EVENT_21
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    PRINTFORML %QST_LEADER_NAME()%はベンチに腰掛けた
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチを立つ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;擬似的に設備:ベンチ(ID=22)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 22
    WAIT
ELSE
    PRINTFORMW %QST_LEADER_NAME()%はベンチを立ち去った……
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
ENDIF

;-------------------------------------------------
;EV30- 教会
;-------------------------------------------------
; #region EV30 教会
;EV30 拝廊、会話イベント
@QST_ADD_EVENT_QUEST_14_EVENT_30
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "拝廊"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_30

IF HO_OCLOCK() < 2200
    PRINTL
    PRINTFORML 教会の入口及び拝廊近辺は、フロリアからの難民と思われる人々でごった返している…
    PRINTFORMW 人々は炊き出しを求めて列を為しているようだ。誰かに話を聞いてみようか？
    $ INPUT_LOOP
    PRINTL
    CALL ASK_MULTI_JUDGE("恰幅の良い女性", 1, "汚い身なりの幼女",1, "訳知りの老人", 1, "やめておく", 1)
    SELECTCASE RESULT
        CASE 0
            PRINTL
            SETCOLOR カラー_オレンジ
            PRINTFORML 「大丈夫、大丈夫だよ！」
            PRINTFORMW 「炊き出しは充分にあるからね！ほら、そこ押さない！」
            RESETCOLOR
            PRINTL
        CASE 1
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「お腹空いたよ…」
            PRINTFORMW 「…いつになったらお家に帰れるの…ママ…」
            RESETCOLOR
            PRINTL
        CASE 2
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「炊き出しを取り仕切っている恰幅の良い女性がいるじゃろう？」
            PRINTFORML 「彼女は元々フロリアの名門貴族、ハート家の使用人をしていたと聞いておる」
            PRINTFORMW 「内戦で奮迅の活躍を見せたハート家も、数年前ついに断絶してしまった…不憫なことじゃよ…」
            RESETCOLOR
            PRINTL
        CASEELSE
            RETURN
    ENDSELECT
ELSE
    PRINTL
    PRINTFORMW 教会の入口及び拝廊近辺で、フロリアからの難民と思われる人々が身を寄せ合っている…
    PRINTFORMW 人々は皆一様に疲れ果てている…、誰かに話を聞いてみようか？
    PRINTL
    CALL ASK_MULTI_JUDGE("恰幅の良い女性", 1, "頬のこけた男性",1, "憤っている女性", 1, "やめておく", 1)
    SELECTCASE RESULT
        CASE 0
            PRINTL
            SETCOLOR カラー_オレンジ
            PRINTFORML 「ん……むにゃ……」
            PRINTFORMW 「……ルミナス嬢様……どうか御無事で……むにゃ……」
            RESETCOLOR
            PRINTL
        CASE 1
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「当たり前のことですが、皆が皆やれ王家だの、自由だの大層な御旗の下に戦っているわけではないのですよ」
            PRINTFORML 「王党派にせよ共和派にせよ、末端の徴募兵の帰属なんてものは、せいぜいこっちの村に住んでいたかあっちの村に住んでいたかという違いでしかありません」
            PRINTFORMW 「徴募といえば聞こえがいいですが、実態は半ば強制のようなものです」
            PRINTL
            PRINTFORML 「……私の故郷を焼いた連中には見知った顔もありましてね、かつての友人であったり商売相手であったり」
            PRINTFORML 「彼らに対する恨みや憎しみが無いと言えば…もちろん嘘になりますが……」
            PRINTFORMW 「それ以上に、まぁなんとも虚しく、やるせないものですよ……。見知った者同士で相争うというのは」
            RESETCOLOR
            PRINTL
            PRINTFORMW そういって男は肩を落として深々とため息をつくと、空を仰ぎ見た
            PRINTL
        CASE 2
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「王党派の貴族の中には、領地の防衛そっちのけでブーン炭鉱結社や教会との利権争いに目が眩んでる恥知らずもいる…」
            PRINTFORML 「民を守るはずの貴族達がそんな体たらくだから、いつまでも戦争が終わらないんだよ…！」
            PRINTFORMW 「それに比べてハート家の方々は本当に立派だったよ…最後まで王国と民に殉じて……」
            RESETCOLOR
            PRINTL
        CASEELSE
            RETURN
    ENDSELECT
ENDIF
GOTO INPUT_LOOP

;EV31 礼拝堂、会話イベント
@QST_ADD_EVENT_QUEST_14_EVENT_31
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "礼拝堂"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_31
IF HO_OCLOCK() < 2200
    PRINTL
    PRINTFORML 礼拝堂では、神父らと共にフロリアからの難民と思われる人々が祈りを捧げている…
    PRINTFORML 熱心に祈りを捧げる神父と難民達に、脇から厳しい視線を注いでいる者達がいる…
    PRINTFORMW どうやら教会の尖兵達のようだが……、この場の誰かに話を聞いてみようか？
    PRINTL
    $ INPUT_LOOP
    CALL ASK_MULTI_JUDGE("神父", 1, "若いテンプルナイト兵卒",1, "壮年のテンプルナイト兵卒", 1, "やめておく", 1)
    SELECTCASE RESULT
        CASE 0
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「近日、教皇庁よりジスト枢機卿が見えられます！」
            PRINTFORML 「今の苦境は決して長くは続きません！枢機卿が皆様と共に祈るのですから…」
            PRINTFORMW 「どうか皆様に神の御加護がありますように…！」
            RESETCOLOR
            PRINTL
        CASE 1
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「お前は難民じゃないんだろ？だったらここに用は無いはずだ」
            PRINTFORML 「…それにしてもジスト枢機卿はなんでこのタイミングで…、まったく面倒なことだ……」
            PRINTFORMW 「…なんだ、まだいたのか？邪魔だからさっさと失せな」
            RESETCOLOR
            PRINTL
        CASE 2
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORMW 「……神聖な祈りの最中だ。私語は慎め」
            RESETCOLOR
            PRINTL
        CASEELSE
            RETURN
    ENDSELECT
ELSE
    PRINTL
    PRINTFORML 祈りの時間を終えた礼拝堂は、年端もいかぬ子ども達の寝息が静かに響き渡っていた
    PRINTFORML 恐らくフロリア難民の子ども達であろう…、眠る場所はせめて暖かいところで、との配慮だろうか？
    PRINTFORMW まだ起きている者も僅かながらいるようだ…、誰かに話を聞いてみようか？
    PRINTL
    CALL ASK_MULTI_JUDGE("神父", 1, "若いテンプルナイト兵卒",1, "壮年のテンプルナイト兵卒", 1, "やめておく", 1)
    SELECTCASE RESULT
        CASE 0
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「残念ながら、私には戦争を止める力も、魔物や野盗に苦しめられる人々を救う力もありません」
            PRINTFORMW 「せめて、子どもたちだけは、この場所で少しでも安らかな夜を迎えてほしいものです」
            PRINTL
            PRINTFORML 「イコンの地の同志も、亜人達を救うことができず、日夜暗闘を続けていると聞きます」
            PRINTFORMW 「…ジスト枢機卿が何かを変えてくれればよいのですが……」
            RESETCOLOR
            PRINTL
        CASE 1
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「はあ、今日もやっと祈りの時間が終わったよ……、神父さんもよく飽きないもんだな」
            PRINTFORMW 「信仰者ってのは皆、こんな人種なのか？俺にはよく分からないな」
            PRINTL
            PRINTFORML 「とにかく俺は、騎士（ナイト）の方々に目を付けられないよう、無難なお勤めを心掛けねえと……」
            PRINTFORMW 「…ジスト卿みたいに、あいつらに目を付けられて、昇格という名の左遷、なんてたまったもんじゃねえ」
            RESETCOLOR
            PRINTL
        CASE 2
            PRINTL
            SETCOLOR カラー_ライトブルー
            PRINTFORML 「フロリアの内戦に対しては表向きはどの勢力も中立を掲げているが、欺瞞もいいところだ」
            PRINTFORML 「どいつもこいつも火が絶えぬよう外様から好き勝手に薪を投げ込んでいきやがるが、
            PRINTFORMW 焼け出された連中のことなんざ気にもかけん」
            PRINTL
            PRINTFORMW 「くくっ、教会の走狗に言えたことじゃあないな……。喋りすぎた。聞かなかったことにしてくれ」
            RESETCOLOR
            PRINTL
        CASEELSE
            RETURN
    ENDSELECT
ENDIF
GOTO INPUT_LOOP

; #endregion

;EV39 教会の裏手
@QST_ADD_EVENT_QUEST_14_EVENT_39
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "教会の裏手に入る"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_14_EVENT_39
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    CALL FL_PAUSEBGM()
    PRINTFORMW %QST_LEADER_NAME()%は人目のつかない教会の裏手に忍び込んだ……
    CALL QST_SET_EVENT_SLOT_NAME(, @"教会の裏手を出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;ウフフ系コマンドの実行を許可
    FLAG:クエスト_ウフフ許可 = 1
ELSE
    CALL FL_REPLAYBGM()
    PRINTFORMW %QST_LEADER_NAME()%は教会の裏手から出た……
    CALL QST_SET_EVENT_SLOT_NAME(, @"教会の裏手に入る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
ENDIF
; #endregion

;-------------------------------------------------
;EV40- 市場
;-------------------------------------------------
; #region EV40 市場
@QST_ADD_EVENT_QUEST_14_EVENT_40
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "露店"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:SHOP]"

@QST_QUEST_14_EVENT_40
CALL HO_SHOP_MENU("QUEST_14_露店", "市場の露店", "QUEST_14_DIC_SHOP_STOCK_0")

@QST_ADD_EVENT_QUEST_14_EVENT_41
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "アクセサリ屋"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:SHOP]"

@QST_QUEST_14_EVENT_41
CALL HO_SHOP_MENU("QUEST_14_アクセサリ屋", "アクセサリ屋", "QUEST_14_DIC_SHOP_STOCK_1")

;EV41 行商人、会話イベント
@QST_ADD_EVENT_QUEST_14_EVENT_42
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "貿易商"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_42
PRINTFORML 「高い……高い……、なんでも高い……」
PRINTFORMW 「あぁ、すまん。しかしお前さんも見ただろうあの鉱物の値段。半年前の倍はする」
PRINTL
PRINTFORML 「鉱物に限らずあらゆる資源は工場とやらがみーんな持っていっちまうから高騰が止まらんのだよ」
PRINTFORMW 「需要に対して供給がまるで追いついとらんというわけだ」

;EV41 行商人、会話イベント2
@QST_ADD_EVENT_QUEST_14_EVENT_43
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "自由都市の交易商"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_43
PRINTFORML 「石炭はノルディアに持っていけさえすればべらぼうに儲かるんだが、いかんせん産地が限られとる」
PRINTFORMW 「南レルムや南方大陸には手つかずの鉱脈が幾つもあるって噂だが、どっちも魔物がうじゃうじゃいるからなぁ」
PRINTFORMW 「良質な天然鉱があれば石炭だろうが硫黄だろうが自前で調達できるんだが……」

;EV44 行商人、会話イベント3
@QST_ADD_EVENT_QUEST_14_EVENT_44
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "隊商の長"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_44
PRINTFORML 「遥か東方への路を往来し、オリエンスの品々を商うキャラバンも、最盛期に比べればめっきり減ったよ」
PRINTFORMW 「ノルディアが東方世界へ船を繰り出すようになってから、東方の産品も随分安く出回るようになったからね」
PRINTL
PRINTFORMW 「まあ、俺たちは俺たちで細々とやっていくさ」

;EV45 食料市
@QST_ADD_EVENT_QUEST_14_EVENT_45
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "食料市"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:SHOP]"

@QST_QUEST_14_EVENT_45
CALL HO_SHOP_MENU("QUEST_14_食料市", "食料市", "QUEST_14_DIC_SHOP_STOCK_2")

; #endregion
;-------------------------------------------------
;EV60- ウォーターフロント
;-------------------------------------------------
;EV60 ウォーターフロントの船乗り、会話イベント
@QST_ADD_EVENT_QUEST_14_EVENT_60
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "船乗り"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_60
PRINTFORMW 「あれが見えるか？」
PRINTL
PRINTFORML 船乗りが指さした先には煙突から黒煙を吐き出して進む船の姿が見える
PRINTFORMW 船の両舷には水車のような機構が取り付けられており、回転させることで推進力を得ているようだ
PRINTL
PRINTFORML 「ノルディアの外輪帆船だ、この辺りで見るのは珍しいな」
PRINTFORMW 「あれに乗れれば季節風の顔色を伺わずに航海ができるんだがなぁ……」

;-------------------------------------------------
;EV100- 南市門
;-------------------------------------------------
;EV110 南市門の衛兵、会話イベント
@QST_ADD_EVENT_QUEST_14_EVENT_100
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "南市門の衛兵"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_100
PRINTFORMW 「この南街道を白霜山脈に沿ってずーっと南に進むとノアレルムに行き着く」
PRINTL
PRINTFORML 「南から運ばれてきた資源は大河から北海を海路伝いにコモンウェルスへ」
PRINTFORMW 「コモンウェルスから運ばれてきた武器や工業品はここから陸路伝いに南へ行くってわけだ」
PRINTL
PRINTFORMW 「フロリアの内乱で西商路が使い物にならんから、この東商路は今ちょっとした大騒ぎってわけよ」
PRINTFORMW 「おおっと、無駄口叩いてる場合じゃなかった……ほら、そこのお前！止まれ止まれ！まだ通行許可は出してないぞ！」

;-------------------------------------------------
;EV110 - スラムのイベント
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_110
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "BSB団のアジト"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:TALK]"

@QST_QUEST_14_EVENT_110
#DIM CONST EVENT_ID = 110
#DIM EVENT_FLAG
PRINTL
PRINTFORML スラム街の一角、ボロボロの掘っ立て小屋の周りを、ガラの悪い若者達がひっきりなしに出入りしている…
PRINTFORMW 誰かに話を聞いてみようか？
PRINTL
CALL ASK_MULTI_JUDGE("アジトの入口", 1, "見張りの男性", 1, "やめておく", 1)
SELECTCASE RESULT
    CASE 0
        PRINTL
        PRINTFORMW アジトの入口の守りは固そうだ…
        ;クエストフラグ立てる予定
        PRINTL
    CASE 1
        PRINTL
        SETCOLOR カラー_ライトブルー
        PRINTFORML 「フォルメンのスラムは俺たち、【バッド・ストリート・ボーイズ】のナワバリだぜ？」
        PRINTFORMW 「みかじめ料を払う気はあるんだろうなあ？……って言いたいところだが…」
        PRINTL
        PRINTFORML 「ジスト枢機卿とかいうのが近々この街を訪れるんだとよ」
        PRINTFORML 「そうでなくったって厳しいテンプルナイト共の締め付けがますます厳しくなっちまったよ」
        PRINTFORMW 「俺らみたいな日陰者は表通りを歩くこともできやしねえ…面白くねえな…」
        RESETCOLOR
        PRINTL
    CASEELSE
        RETURN
ENDSELECT

;-------------------------------------------------
;EV120 - スラムのイベント
;-------------------------------------------------
;-------------------------------------------------
;EV120 裏路地の小商店
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_120
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "路地裏の小商店"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_14_EVENT_120
CALL HO_SHOP_MENU("QUEST_14_路地裏の小商店", "路地裏の小商店")

;怪しい売人イベント
@QST_ADD_EVENT_QUEST_14_EVENT_123
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "怪しい売人"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHARM]"

; ゲーム中一度でも解決済みの場合は非アクティブに
; CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
; IF RESULT
;     CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
; ENDIF

@QST_QUEST_14_EVENT_123
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC MOB_ID
#DIM DYNAMIC 性具絶頂回数
#DIM DYNAMIC SEDU_COUNT

MOB_ID = TOINT(QST_GET_FLAGMAP("EVENT_123_MOB_ID"))
;イベント未開始
IF !MOB_ID
    PRINTFORML 路地裏のいかがわしい商店の前に売人と思しき男が立っている
    PRINTFORMW 目が合うと男は舐めるような視線で、品定めするように%QST_LEADER_NAME()%を見た
    PRINTL
    FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
        CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
        IF IS_FEMALE(CHARA_ID) && HAS_SKILL(CHARA_ID, "SEDUCTION", スキル_大分類_性技)
            SEDU_COUNT ++
            BREAK
        ENDIF
    NEXT

    IF SEDU_COUNT && HO_OCLOCK() < 1400
        PRINTFORML 「よお、%IS_MALE(QST_LEADER_ID()) ? "旦那" # "お嬢さん"%、ちょっとした"玩具"に興味は無いかい？」
        PRINTFORMW 「新しい商品を開発したんだが、ちょいと試してみたくてね……」
        PRINTL
        CALL ASK_YN("興味がある", "興味ない")
        IF RESULT
            PRINTFORMW 「そうかいそうかい、まぁ興味があったら声をかけてくれよ」
            RETURN
        ENDIF
        PRINTFORML 「よしきた、こいつはちょっとした震える玩具で、手を動かさなくても勝手に気持ちよくなれる優れモノだ」
        PRINTFORMW 「俺の見立てじゃぁ……あんたはきっと気に入ってくれると思ってね」
        PRINTFORMW 男の下卑た視線が%ANAME(CHARA_ID)%に向く
        PRINTL
        PRINTFORMW 「もしテストに付き合ってくれたら、こいつをくれてやろう」
        PRINTFORMW そう言って男はふるふると振動する小石ほどの性具を取り出すと、にたりといやらしい笑みを浮かべた……
        PRINTL
        CALL PRINT_TITLE("震える試供品", "エロイベント")
        CALL QST_SELECT_CHARA(,,"QST_CHARA_IS_SEDUCTIVE_FEMALE")
        CHARA_ID = RESULT
        IF CHARA_ID == -1
            PRINTFORMW 「そうかいそうかい、まぁ興味があったら声をかけてくれよ」
            RETURN
        ENDIF
        ;使用していないエリアにダミーキャラを生成
        CALL HO_CREATE_GUEST(24, "DUMMY", "商人", 0)
        MOB_ID = RESULT
        ABL:MOB_ID:性技 = 15
        ABL:MOB_ID:奉仕 = 15
        TALENT:MOB_ID:技師 = 1
        PRINTL
        PRINTFORMW 「くくっ、やはり見立ての通りだ。ほらよ」
        IF !HAS_VAGINA(CHARA_ID)
            PRINTFORMW 受け取ってはみたものの、どうやらこれは女性器用の性具のようだ
            PRINTL
            PRINTFORML 「どうした、やっぱりやめるなんて言わないでくれよ」
            PRINTFORMW 「わかったわかった、そいつがどうしても嫌なら、こいつはどうだ」
            PRINTFORMW %ANAME(CHARA_ID)%の様子を見て何を勘違いをしたのか、別の性具と交換された
            PRINTL
            PRINTFORMW 「これなら生娘のあんたでも気に入ると思うぜ」
            PRINTFORMW 振動する小石がついた、キャップ型の小さな性具を２つほど手の中に押し込まれた……
        ENDIF
        PRINTL
        ;ダミーモブを実行者としてローターコマンドを継続ロック状態でセット
        CALL CLEAR_MPLY()
        CALL CLEAR_MTAR()
        CALL ADD_MPLY(MOB_ID)
        CALL ADD_MTAR(CHARA_ID)
        IF HAS_VAGINA(CHARA_ID)
            CALL SET_EQUIP(T_N("ローター"), 1, 1)
            CALL QST_SET_FLAGMAP("EVENT_123_性具絶頂回数", TOSTR(EX:CHARA_ID:Ｃ絶頂))
        ELSE
            CALL SET_EQUIP(T_N("ニプルキャップ"), 1, 1)
            CALL QST_SET_FLAGMAP("EVENT_123_性具絶頂回数", TOSTR(EX:CHARA_ID:Ｂ絶頂))
        ENDIF
        CALL QST_SET_FLAGMAP("EVENT_123_MOB_ID", TOSTR(MOB_ID))
        CALL QST_SET_FLAGMAP("EVENT_123_CHARA_ID", TOSTR(CHARA_ID))
        CALL RESET_ACTOR()
        CALLF TAG_PRINT (@"@C:{カラー_ピンク}@%ANAME(CHARA_ID)%は受け取った性具を物陰でそっと\@ HAS_VAGINA(CHARA_ID) ? 秘部 # 胸の先端 \@にあてがった……@/C@", 1)
        WAIT
        CALLF TAG_PRINT (@"「一つ条件がある。特に耐久性が確かめたいんでなぁ。@C:{カラー_注意}@19時@/C@までは付けっぱなしにしててもらいたい」", 1)
        PRINTFORMW 「時間になったらまた話しかけてくれ。お前さんがきちんとこいつを付けたままならこいつといくらか報酬をやろう」
        PRINTL
        PRINTFORMW 「……そのままくすねようとか変なことは考えるなよ？　こいつは外れたら俺の手元に戻ってくる。盗難対策はバッチリってわけだ」
        PRINTFORMW 「それじゃあまぁ……せいぜい楽しんでくれ」
    ELSEIF SEDU_COUNT && HO_OCLOCK() >= 1400
        PRINTL
        PRINTFORMW 「……ふーむ、お前さんみたいなやつに頼みたかった依頼があるんだがもう少し早い時間ならなぁ……」
        PRINTFORMW 「まぁ興味があるなら昼前にでも声をかけてくれよ」
    ELSE
        ;条件に合致するキャラがパーティにいない
        PRINTFORMW ……しかし、すぐに興味を失ったのか、目をそらした
    ENDIF
    ;イベント開始済
ELSE
    CHARA_ID = TOINT(QST_GET_FLAGMAP("EVENT_123_CHARA_ID"))
    IF IS_EQUIP_PLAYER(MOB_ID, T_N("ローター")) || IS_EQUIP_PLAYER(MOB_ID, T_N("ニプルキャップ"))
        IF HO_OCLOCK() >= 1900
            PRINTFORMW 「ほう……くくっ、きちんと付けたまま過ごしてきたようだな……どれどれ」
            IF HAS_VAGINA(CHARA_ID)
                性具絶頂回数 = EX:CHARA_ID:Ｃ絶頂 - TOINT(QST_GET_FLAGMAP("EVENT_123_性具絶頂回数"))
            ELSE
                性具絶頂回数 = EX:CHARA_ID:Ｂ絶頂 - TOINT(QST_GET_FLAGMAP("EVENT_123_性具絶頂回数"))
            ENDIF
            IF 性具絶頂回数 > 10
                PRINTFORMW 「ははっ……！　{性具絶頂回数}回も達したのか！　大したもんだ、大いに楽しんでたようで何よりだぜ！」
            ELSEIF 性具絶頂回数 > 1
                PRINTFORMW 「ほう……{性具絶頂回数}回もイったのか。　なかなか楽しんでくれたようだな」
            ELSEIF 性具絶頂回数 == 1
                PRINTFORMW 「その様子を見る限りまぁ少しは楽しんでくれたようだな」
            ELSE
                PRINTFORMW 「ふーむ、性能の問題か？　お前さんはあまりお気に召さなかったみたいだな」
            ENDIF
            PRINTL
            IF HAS_VAGINA(CHARA_ID)
                PRINTFORMW 「……あぁ、約束通りそいつはくれてやるよ。気に入ってくれたならどんどん使ってくれよな」
                PRINTL
                CALL QST_CHANGE_ITEM(@"{ITEM_ローター},1,淫堕点,25,MONEY,4500,")
            ELSE
                PRINTFORMW 「……あぁ、約束通りそいつはくれてやるよ。気に入ってくれたならどんどん使ってくれよな」
                PRINTL
                CALL QST_CHANGE_ITEM(@"{ITEM_ニプルキャップ},1,淫堕点,25,MONEY,4500,")
            ENDIF
            CALL RELEASE_ALL_EQUIP(MOB_ID)
            ; CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
            CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
        ELSE
            PRINTFORMW 「なんだ？　まだ約束の時間は来てないぞ」
            PRINTFORMW 「19時まではそいつを付けたまま街で過ごしてきてくれよ。どう楽しむかは……おまえさんに任せるがね、くくっ……」
        ENDIF
    ELSE
        PRINTFORMW 「なんだ、外しちまったのか。やれやれ、これじゃあ今日はもうテストにならんな」
        PRINTFORMW 「まぁ……また次来た時やる気があるなら声をかけてくれ」
        CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
    ENDIF
ENDIF

;-------------------------------------------------
;EV301 路地裏
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_301
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "路地裏に入る"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_14_EVENT_301
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    CALL FL_PAUSEBGM()
    PRINTFORMW %QST_LEADER_NAME()%は人目のつかない路地裏に忍び込んだ……
    CALL QST_SET_EVENT_SLOT_NAME(, @"路地裏を出る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;ウフフ系コマンドの実行を許可
    FLAG:クエスト_ウフフ許可 = 1
    ;擬似的に設備:路地裏(ID=900)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 900
ELSE
    CALL FL_REPLAYBGM()
    PRINTFORMW %QST_LEADER_NAME()%は路地裏から出た……
    CALL QST_SET_EVENT_SLOT_NAME(, @"路地裏に入る")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
ENDIF

;-------------------------------------------------
;EV302 ベンチ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_14_EVENT_302
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "ベンチ"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:3 '= "[EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_14_EVENT_302
#DIM EVENT_FLAG
#DIM CHARA_ID

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

IF !EVENT_FLAG
    PRINTFORML %QST_LEADER_NAME()%はベンチに腰掛けた
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチを立つ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 1)
    ;このイベントIDのイベント以外実行不可に
    QST_イベント選択制限 = QST_NOWEV
    ;1=隣接エリアのみ移動可,2=エリア移動不可
    FLAG:移動制限フラグ = 2
    ;擬似的に設備:ベンチ(ID=22)を使っている状態にする
    TFLAG:クエスト_ダミー設備ID = 22
    ;夜ならウフフ系コマンドの実行を許可
    IF HO_OCLOCK() >= 2000
        FLAG:クエスト_ウフフ許可 = 1
        PRINTFORML 今は辺りに誰も居ないようだ……
    ENDIF
    WAIT
ELSE
    PRINTFORMW %QST_LEADER_NAME()%はベンチを立ち去った……
    CALL QST_SET_EVENT_SLOT_NAME(, @"ベンチ")
    CALL QST_SET_EVENT_FLAG(QST_NOWEV, 0)
    QST_イベント選択制限 = -1
    FLAG:移動制限フラグ = 0
    FLAG:クエスト_ウフフ許可 = 0
    TFLAG:クエスト_ダミー設備ID = 0
    CALL QST_REDRESSALL
ENDIF

;-------------------------------------------------
;ショップの定義
;関数名は@HO_SHOP_MANAGE_SHOP_{ショップ名称}で定義
;【メモ】
;露店は在庫データをセーブするために専用の広域変数を確保している。
;在庫をセーブに保持する必要がないならDICS_SHOP_STOCK:nに直接販売したいアイテムの情報を書き込めば良い
;-------------------------------------------------
@HO_SHOP_MANAGE_SHOP_QUEST_14_露店(DICS_SHOP_STOCK)
#DIMS REF DICS_SHOP_STOCK
IF QUEST_14_DIC_SHOP_STOCK_0:0 == ""
    QUEST_14_DIC_SHOP_STOCK_0:0 '= @"[ITEM_ID:{ITEM_硫黄}][ITEM_PRICE:2000]"
    QUEST_14_DIC_SHOP_STOCK_0:1 '= @"[ITEM_ID:{ITEM_石炭}][ITEM_PRICE:1600]"
    QUEST_14_DIC_SHOP_STOCK_0:2 '= @"[ITEM_ID:{ITEM_銀インゴット}][ITEM_PRICE:12000]"
    {
        QUEST_14_DIC_SHOP_STOCK_0:3 '= @"
            [WEAPON_ID:1][EQUIP_TYPE:1][WEAPON_NAME:赤べこ]
            [WEAPON_RANK:3][MATERIAL:{WPN_材質_木}]
            [ITEM_PRICE:4000][SHOP_STOCK:5]
            "
    }
    {
        QUEST_14_DIC_SHOP_STOCK_0:4 '= @"
            [WEAPON_ID:2][EQUIP_TYPE:1][WEAPON_NAME:木彫りのタリズマン]
            [WEAPON_RANK:0][MATERIAL:{WPN_材質_木}]
            [ITEM_PRICE:10000][SHOP_STOCK:2]
            "
    }
    {
        QUEST_14_DIC_SHOP_STOCK_0:6 '= @"
            [WEAPON_ID:1][EQUIP_TYPE:0][WEAPON_NAME:盗賊の短刀]
            [WEAPON_RANK:0][MATERIAL:{WPN_材質_鋼}]
            [PASSIVE_WEAPON_SKILL:WPN_ADDITIONAL_ATTACK,1,]
            [ITEM_PRICE:25000][SHOP_STOCK:1]
            "
    }
ENDIF
ARRAYCOPY "QUEST_14_DIC_SHOP_STOCK_0", "DICS_SHOP_STOCK"

@HO_SHOP_MANAGE_SHOP_QUEST_14_アクセサリ屋(DICS_SHOP_STOCK)
#DIMS REF DICS_SHOP_STOCK
IF QUEST_14_DIC_SHOP_STOCK_1:0 == ""
    {
        QUEST_14_DIC_SHOP_STOCK_1:0 '= @""
            + @"[WEAPON_ID:2][EQUIP_TYPE:1][WEAPON_NAME:抗堪の守り]"
            + @"[WEAPON_RANK:0][MATERIAL:{WPN_材質_鉄}]"
            + @"[PASSIVE_WEAPON_SKILL:WPN_BOOST_REGIST_STUN,3,]"
            + @"[ITEM_PRICE:35000][SHOP_STOCK:3]"
    }
    {
        QUEST_14_DIC_SHOP_STOCK_1:1 '= @""
            + @"[WEAPON_ID:2][EQUIP_TYPE:1][WEAPON_NAME:抗毒の守り]"
            + @"[WEAPON_RANK:0][MATERIAL:{WPN_材質_鉄}]"
            + @"[PASSIVE_WEAPON_SKILL:WPN_BOOST_REGIST_POISON,3,]"
            + @"[ITEM_PRICE:35000][SHOP_STOCK:3]"
    }
    {
        QUEST_14_DIC_SHOP_STOCK_1:2 '= @""
            + @"[WEAPON_ID:2][EQUIP_TYPE:1][WEAPON_NAME:耐火の守り]"
            + @"[WEAPON_RANK:0][MATERIAL:{WPN_材質_鉄}]"
            + @"[PASSIVE_WEAPON_SKILL:WPN_BOOST_REGIST_BURN,3,]"
            + @"[ITEM_PRICE:35000][SHOP_STOCK:3]"
    }
    {
        QUEST_14_DIC_SHOP_STOCK_1:3 '= @""
            + @"[WEAPON_ID:6][EQUIP_TYPE:1][WEAPON_NAME:古風なモノクル]"
            + @"[WEAPON_RANK:0][MATERIAL:{WPN_材質_銀}]"
            + @"[PASSIVE_WEAPON_SKILL:WPN_MAG_PENETRATE,1,WPN_MAG_BONUS,2,]"
            + @"[ITEM_PRICE:50000][SHOP_STOCK:1]"
    }
ENDIF
ARRAYCOPY "QUEST_14_DIC_SHOP_STOCK_1", "DICS_SHOP_STOCK"

@HO_SHOP_MANAGE_SHOP_QUEST_14_路地裏の小商店(DICS_SHOP_STOCK)
#DIMS REF DICS_SHOP_STOCK

DICS_SHOP_STOCK:0 '= @"[ITEM_ID:{ITEM_媚薬}][ITEM_PRICE:4000]"
DICS_SHOP_STOCK:1 '= @"[ITEM_ID:{ITEM_鞭}][ITEM_PRICE:6000]"
DICS_SHOP_STOCK:2 '= @"[ITEM_ID:{ITEM_オナホール}][ITEM_PRICE:8000]"
DICS_SHOP_STOCK:3 '= @"[ITEM_ID:{ITEM_ローター}][ITEM_PRICE:10000]"
DICS_SHOP_STOCK:4 '= @"[ITEM_ID:{ITEM_バイブ}][ITEM_PRICE:14000]"
DICS_SHOP_STOCK:5 '= @"[ITEM_ID:{ITEM_アナルバイブ}][ITEM_PRICE:14000]"
{
    DICS_SHOP_STOCK:6 '= @"
        [WEAPON_ID:20][EQUIP_TYPE:1][WEAPON_NAME:汚れた盾の呪印]
        [WEAPON_RANK:0][MATERIAL:{WPN_材質_銀}]
        [PASSIVE_WEAPON_SKILL:WPN_VAGINA_GUARD,1,]
        [ITEM_PRICE:30000]
        "
}
{
    DICS_SHOP_STOCK:7 '= @"
        [WEAPON_ID:20][EQUIP_TYPE:1][WEAPON_NAME:朽ちた剣の呪印]
        [WEAPON_RANK:0][MATERIAL:{WPN_材質_銀}]
        [PASSIVE_WEAPON_SKILL:WPN_PENIS_GUARD,1,]
        [ITEM_PRICE:30000]
        "
}

@HO_SHOP_MANAGE_SHOP_QUEST_14_食料市(DICS_SHOP_STOCK)
#DIMS REF DICS_SHOP_STOCK
IF QUEST_14_DIC_SHOP_STOCK_2:0 == ""
    SELECTCASE HO_SEASON()
        CASE 0
            QUEST_14_DIC_SHOP_STOCK_2:0 '= @"[ITEM_ID:{ITEM_キャベツ}][ITEM_PRICE:500]"
            QUEST_14_DIC_SHOP_STOCK_2:1 '= @"[ITEM_ID:{ITEM_ニンニク}][ITEM_PRICE:600]"
        CASE 1
            QUEST_14_DIC_SHOP_STOCK_2:0 '= @"[ITEM_ID:{ITEM_トマト}][ITEM_PRICE:600]"
            QUEST_14_DIC_SHOP_STOCK_2:1 '= @"[ITEM_ID:{ITEM_カボチャ}][ITEM_PRICE:600]"
        CASE 2
            QUEST_14_DIC_SHOP_STOCK_2:0 '= @"[ITEM_ID:{ITEM_ニンジン}][ITEM_PRICE:600]"
            QUEST_14_DIC_SHOP_STOCK_2:1 '= @"[ITEM_ID:{ITEM_タマネギ}][ITEM_PRICE:400]"
        CASE 3
            QUEST_14_DIC_SHOP_STOCK_2:0 '= @"[ITEM_ID:{ITEM_カブ}][ITEM_PRICE:500]"
            QUEST_14_DIC_SHOP_STOCK_2:1 '= @"[ITEM_ID:{ITEM_ネギ}][ITEM_PRICE:600]"
    ENDSELECT
    QUEST_14_DIC_SHOP_STOCK_2:2 '= @"[ITEM_ID:{ITEM_レッドペッパー}][ITEM_PRICE:900]"
    QUEST_14_DIC_SHOP_STOCK_2:3 '= @"[ITEM_ID:{ITEM_岩塩}][ITEM_PRICE:500]"
    QUEST_14_DIC_SHOP_STOCK_2:4 '= @"[ITEM_ID:{ITEM_バター}][ITEM_PRICE:800][SHOP_STOCK:50]"
    QUEST_14_DIC_SHOP_STOCK_2:5 '= @"[ITEM_ID:{ITEM_パスタ}][ITEM_PRICE:1000][SHOP_STOCK:20]"
    QUEST_14_DIC_SHOP_STOCK_2:6 '= @"[ITEM_ID:{ITEM_チーズ}][ITEM_PRICE:1600][SHOP_STOCK:30]"
    QUEST_14_DIC_SHOP_STOCK_2:7 '= @"[ITEM_ID:{ITEM_川エビ}][ITEM_PRICE:400]"
    QUEST_14_DIC_SHOP_STOCK_2:8 '= @"[ITEM_ID:{ITEM_干し肉}][ITEM_PRICE:3300][SHOP_STOCK:20]"
    QUEST_14_DIC_SHOP_STOCK_2:9 '= @"[ITEM_ID:{ITEM_東方スパイス}][ITEM_PRICE:30000][SHOP_STOCK:10]"
ENDIF
ARRAYCOPY "QUEST_14_DIC_SHOP_STOCK_2", "DICS_SHOP_STOCK"

path: ERB/QUEST/クエストデータ/QUEST_14_交易都市フォルメン/QUEST_14_交易都市フォルメン.ERB